


# TVORIVÁ INFORMATIKA

## 1. zošit z programovania



Slovenské pedagogické nakladateľstvo



# OBSAH


	<b>Krajina Imagine</b>	<b>4</b>
<b>1</b>	<b>Pero, farba a hrúbka</b>	<b>6</b>
<b>2</b>	<b>Opakuj a pomenuj</b>	<b>12</b>
<b>3</b>	<b>Stavebnice príkazov</b>	<b>16</b>
<b>4</b>	<b>Dôležité udalosti v živote korytnačky</b>	<b>20</b>
<b>5</b>	<b>Viac korytnačiek, viac možností</b>	<b>24</b>
<b>6</b>	<b>Korytnačky a ich tvary</b>	<b>28</b>
<b>7</b>	<b>Animované tvary a procesy</b>	<b>32</b>
<b>8</b>	<b>Príkazy s premennými</b>	<b>36</b>
<b>9</b>	<b>Pohyby a preteky</b>	<b>40</b>
<b>10</b>	<b>Pokusy a ďalšie hry</b>	<b>44</b>
	<b>Čo sme sa naučili</b>	<b>48</b>

# KRAJINA IMAGINE



Nájdí na pracovnej ploche svojho počítača takúto dvojicu ikon – korytnáčiek. Klikni myšou na ľavú z nich, teda na celú zelenú a stlač kláves **Enter**. Tak vstúpiš do programu, ktorý sa nazýva **Imagine**. Vznikol pre žiakov, študentov a ich učiteľov, aby sa učili skúmať a objavovať pomocou počítača.

Tvojím priateľom a sprievodcom bude korytnačka – malý **kresliaci robot**, ktorý žije na bielej stránke v počítači. Korytnačka sedí a čaká, kedy sa bude podľa tvojich príkazov hýbať a otáčať, akým perom bude kresliť, kedy zotrie papier a začne odznova.



S korytnačkou sa budeme *rozprávať* pomocou **programovacieho jazyka Logo**. Naučíme sa, akým *slovám* a *vetám* rozumie a akým nie. Spoznáme jazyk, ktorým sa dorozumievame v krajine Imagine, v krajine nápadov a objavov, výskumov, dobrodružstiev a radosti z vlastných úspechov.

Môžeme si myslieť, že krajina Imagine je iba list papiera, na ktorom žije korytnačka. Potom si však všimneme, že táto korytnačka je celkom šikovná a poslušná – bude sa pohybovať a kresliť presne podľa našich príkazov. Čoskoro prídeme aj na to, aká je učenlivá, ako veľa ju môžeme naučiť a čo všetko sa pritom naučíme my.



Uvidíš, že v kresbách sa často opakuje ten istý prvok (ten istý fah či pohyb). Zistíš tiež, ako korytnačku naučiť **nové príkazy**.

Naučíš sa **skladať** jednoduché príkazy do zložitejších. Nakreslíš štvorec, trojuholník, kružnicu... Budeš korytnačke vypínať a zapínať pero.


Zistíš, že korytnačka pozná rôzne **udalosti** a **reaguje na ne** podľa svojho návodu. Takouto udalosťou je napríklad **prikliknutie** alebo **príťahanie** korytnačky.




Vytvoríš **niekoľko ďalších korytnačiek**. Jednu naučíš pri fahaní kresliť obyčajné hviezdy, druhú neobyčajné... a tretiu odlišovať **prikliknutím** na stránku zábavný domček.



Korytnačkám zmeníš tvary na hlavy detí. Tie sa potom budú ukazovať a skrývať v okenách domčekov. Naučíš sa správne ich **oslovovať** a zadávať im príkazy.




Tvarom korytnačky môže byť aj obrázok, ktorý na stránke **žije** – napr. hýbe rukami, lieta či skáče. Z takýchto postavičiek si poskladáš prvú vlastnú hru.



Naučíš sa definovať vlastné **príkazy s premennou**, napríklad **strom** s premennou výškou, **štvorec** s premennou farbou či hrúbkou pera. Vytvoríš si vlastnú maľovanku.




Zostavíš na stránke živý obraz z korytnačiek-loďí, korytnačiek-domov a korytnačiek-balónov. Naučíš ich sledovať farbu na obrázku pod nimi a podľa toho sa ďalej **rozhodovať**.



Vytvoríš niekoľko jednoduchých hier s korytnačkami-písmenami. Naučíš korytnačku zložitejšie sa rozhodovať a reagovať rôznymi spôsobmi.





Najtenšie pero má hrúbku 1, hrubšie 2, potom 3... Od hrúbky 5 už má čiara oblé konce. Zaujímavé kresby vznikajú aj tak, že použijeme veľmi hrubé pero, napr. **nechHp 50** alebo **nechHp 100**. Ak chceme, aby nám poradil Imagine, napíšme iba **nechHp** a stlačme kláves **F9**:

? **nechHp + F9**

Vtedy sa otvorí **okno-pomôcka Hrúbky čiary**. Vyberme si hrúbku a klikneme na tlačidlo **Urob!**.

Nezabudnime, že vždy sa môžeme vrátiť k základnej hrúbke pera pomocou príkazu:

? **nechHp 1**

### ÚLOHA 1


Postupne striedaj dve rôzne hrúbky pera a nakresli takéto alebo podobné obrázky.



### O farbe pera

Okrem rôznych hrúbok môže mať pero korytnačky aj rôzne farby. Ak chceme zmeniť farbu pera, použijeme ďalšie **okno-pomôcku Farby**. Do príkazového riadka napíšme príkaz **nechFarbaPera**, alebo skráteno **nechFp** a stlačme kláves **F9**:

? **nechFp + F9**



V pomôcke **Farby** klikneme na niektorú farbu. Ak chceme, posúvačom nastavíme jej **jas** a klikneme na tlačidlo **Urob!** Potom zadávame ďalšie korytnačie príkazy, ako **do 100** alebo **vp 45**. Ak si už pamätáme mená niektorých farieb, namiesto klávesu **F9** napíšme **medzeru**, **úvodzovku** a **meno farby**.

? **nechFp "olivová**

? **nechFp "olivová4**

# 1 PERO, FARBA A HRÚBKKA



V okne farieb ponúka Imagine tieto farby. Ich mená môžeme používať buď bez čísla na konci, alebo s číselnou príponou od 1 do 12.

## ÚLOHA 2


Skúmajte takéto farebné kresby, skúšajte rôzne hrúbky pera, farby a otáčania korytnačky príkazmi `vp 90` a `vl 90`.

```
? nechHp 11
? nechFp "modrá"
? do 100
? vp 90
? do 20
? vp 90
? nechFp "tehlová"
? do 100
? vl 90
? do 20
? vl 90
? ...
```



## ÚLOHA 3

Niekedy vystačia na kreslenie zaujímavých obrázkov dve farby, z ktorých jedna je biela. Na nasledujúcom obrázku sme zvolili farbu oranžová5 a s hrúbkou 30 sme nakreslili `do 60`. Potom sme zmenili farbu na bielu, hrúbku na 5 a nakreslili `vz 60`. Nakreslite aj iné podobné obrázky.



## Náhodné farby a hrúbky

Príkaz `nechFp` vyžaduje, aby sme označili niektorú farbu, napríklad `nechFp "modrá8"`. Príkaz `nechHp` zasa vyžaduje niektorú hrúbku. Imagine však pozná aj osobitné označenie pre **ľubovoľnú farbu** alebo **ľubovoľnú hrúbku**:

```
nechFp ? - nastavme ľubovoľnú farbu
nechHp ? - nastavme ľubovoľnú hrúbku
```

Skúsme toto:



```
? nechHp ?
? nechFp ?
? do 100
? vp 90
? nechHp ?
? nechFp ?
? do 90
? vp 90
? ...
```

## Farebné bodky v obrázkoch

Naučme sa ďalší zaujímavý príkaz. Nazýva sa `bod`. Za ním napíšme, aký veľký bod má korytnačka nakresliť – takou farbou pera, akú má práve zvolenú.



## ÚLOHA 4

Nakreslite nasledujúci obrázok. Začnite takto:

```
? vp 90
? nechFp "oranžová1"
? bod 50
? do 35
? nechFp "oranžová2"
? ...
```



## ÚLOHA

5

Pomiešali sa nám tri programy a tri obrázky a teraz nevieme, ktorý z nich nakreslí ktorý. Pozorne si prečítaj programy a prezri obrázky. Rozhodni, čo k čomu patri.


? nechFp "fialová	? nechFp "čierna	? nechFp "červená5
? nechHp 17	? nechHp 5	? bod 100
? do 70	? do 50	? nechFp "žltá8
? vz 35	? nechHp 11	? bod 70
? vp 90	? nechFp "červená	
? do 40	? vp 30	
? vl 90	? do 50	
? do 35	? vl 120	
? vz 70	? do 50	
	? vl 120	
	? do 50	



## ÚLOHA

6

Chceme, aby korytnačka nakreslila takéto obrázky. Pozorne si ich prezri a napíš vhodné príkazy.




## Ďalšie pomôcky

Imagine nám aj pri niektorých ďalších príkazoch pomôže zadať vhodnú farbu, hrúbku, uhol, vzdialenosť a pod. Napíšme meno príkazu, napríklad **vp** a stlačme kláves **F9**. Otvorí sa **okno-pomôcka** **Otočenie**. Kompas, ktorý uvidíme, je otočený v smere korytnačky. Chyťme myšou červenú ihlu a otočme ju niekam inam. Potom klikneme na **Urobi!**. Nájďme aj ďalšie príkazy, ktoré majú svoje pomôcky.



## ZAUJÍMAVÝ EXPERIMENT



? vp 17  
? nechFp ?  
? nechHp ?  
? do 200

? do 200

? do 600

? vl 35  
? nechFp ?  
? nechHp ?  
? do 7300

Otočme korytnačku o kúsok vpravo, napríklad **vp 17** a zadajme jej príkaz **do 200**, aby sa priblížila k okraju stránky. Teraz jej opäť zadajme príkaz **do 200**. Odíde zo stránky preč? Nie, ale objaví sa na jej opačnom okraji a pokračuje v kreslení čiary. Preto nakreslenie **dlhšej čiary** vyzerá tak zvláštne. Skúsme teraz **vl 35** a **do 7300**.






## ZAPAMÄTAJME SI

Aj ďalšie príkazy poznajú **čarovný otáznik** – náhodný počet krokov pre korytnačku, náhodné otočenie...

- **nechHp ?** – korytnačka, nastav si nejakú náhodnú hrúbku pera,
- **nechFp ?** – korytnačka, nastav si nejakú náhodnú farbu pera,
- **dopredu ?** – prejdi dopredu o nejaký počet krokov,
- **vzad ?** – choď vzad o nejaký počet krokov,
- **vpravo ?** – otoč sa vpravo o nejaký uhol,
- **vľavo ?** – otoč sa vľavo o nejaký uhol,
- **bod ?** – nakresli bod nejakaj veľkosti.



Opakujeme v príkazovom riadku znovu a znovu takéto príkazy: zvol náhodnú farbu, nakresli náhodne veľký bod, choď dopredu o ? a otoč sa vpravo o ?

## PROJEKT CD

- Klikneme na tlačidlo **Otvoriť projekt** a v okne možnosti **označíme projekt** s názvom **1pero**. Potom klikneme na tlačidlo **Otvor** a sme dnu. Teraz klikajme na tlačidlá, meňme farbu a hrúbku pera, otáčajme a posúvajte korytnačku dopredu. Čo všetko sa dá nakresliť takýmito jednoduchými nástrojmi?




- Všimnime si tlačidlo **zapamätaj si**. Keď niečo nakreslíme a klikneme na toto tlačidlo, korytnačka si zapamätá všetky kroky od posledného **znovu**. Pomocou **znovu vykonaj** ich všetky zopakuje.
- Všimnime si tiež, že korytnačka si svoje príkazy zapisuje do stĺpca medzi tlačidlami a plochou. Vedel by si tento jej záznam prečítať?

znovu	nechFP "modrá bod 50
dopredu 50	nechHP 6 do 50
vpravo 45	bod 25 vp 90
vľavo 30	nechFP "červená do 50
nechFp "červená	bod 50 nechHP 6
nechFp "modrá	do 50 bod 25
nechFp "čierna	vp 90 nechFP "modrá
nechHp 3	do 50 bod 50
nechHp 6	nechHP 6 do 50
nechHp 12	bod 25 vp 90
bod 25	nechFP "červená do 50
bod 50	bod 50 nechHP 6
	do 50 bod 25
zapamätaj si	vp 90 nechFP "modrá
znovu vykonaj	do 50

**ÚLOHA**  
7





Určite sa ti v projekte **1pero** podarilo nakresliť veľa rôznych obrázkov. Dokážeš vytvoriť aj takéto zložité? Presne takéto, možno iba podobné, možno iné, možno ešte ťažšie. Potráp sa.



## Vytvoríme si vlastné tlačidlo

Chceme zmazať celú stránku jediným kliknutím myši? Vytvoríme si vlastné tlačidlo, ktoré pri kliknutí vykoná príkaz **znovu**.


- 1 Klikneme myšou na nástroj **Nové tlačidlo**. Kurzor myši sa zmení na tieň tlačidla.
- 2 Potom klikneme do stránky – na tomto mieste vznikne naše vlastné tlačidlo s nápisom **t1**, čiže **tlačidlo číslo 1**.
- 3 Keď na nové tlačidlo klikneme myšou, nič sa nestane, pretože sme ho ešte *nenaučili*.
- 4 Klikneme teda na svoje nové tlačidlo pravým tlačidlom myši a v ponuke príkazov zvolíme prvý riadok – **Zmeň t1**.
- 5 V okne vyplníme do riadka **prízapnutí** príkaz **znovu**. Ten sa vykoná vždy, keď na toto tlačidlo klikneme myšou.
- 6 Okno zavrieme tlačidlom **OK**.



## ÚLOHA

8

Práve sme si vytvorili tlačidlo na zmazanie stránky. Rovnako teraz vytvor ďalšie dve tlačidlá. Jedno nech pri každom zapnutí vykoná príkaz **do 50** a nech popis na ňom je rovnaký – **do 50**. Druhé nech vykoná **vp 90** a popis na ňom nech je tiež **vp 90**. Nastav hrúbku pera **9** a farbu **červená** a iba pomocou tlačidiel nakresli:



## ČO SME SA NAUČILI

- Korytnačke môžeme zadávať príkazy, musia však byť presne v takom tvare, ako je zvyknutá.
- Najjednoduchšie je zadávať príkazy **dopredu** alebo **vpravo** a **vľavo**. Každý z nich vyžaduje, aby sme povedali, **o koľko krokov** má ísť korytnačka dopredu alebo **o aký uhol** sa má otočiť.
- Okrem dlhých slov rozumie korytnačka aj niektorým skratkám: **do** znamená **dopredu**, **vp** zasa **vpravo**...
- Lahko môžeme zmeniť farbu alebo hrúbku pera, napríklad pomocou **pomôcok**.
- Stránku zmažeme príkazom **znovu**. Vieme si vytvoriť vlastné tlačidlo, ktoré to urobí za nás.



Otvorme Imagine a opäť si do rohu stránky vytvoríme tlačidlo **Znovu**, ktoré pri zapnutí vykoná príkaz **znovu**. Pri svojich pokusoch ho často použijeme. Korytnačke budeme teraz dávať také príkazy, aby nakreslila farebný balónik. Najprv nastavme hrúbku pera:

```
? nechHp 2
? nechFp ?
? do 120
? bod 60
```



Zvolíme náhodnú farbu pera. Príkážeme korytnačke, aby nakreslila motúz dĺžky 120. Potom jej prikážeme, aby nakreslila veľký bod – balón.

```
? nechHp 2
? nechFp ? do 120 bod 60
```



Kliknime na tlačidlo **Znovu**. Hrúbku pera už meniť nemusíme. Ale nastavenie farby pera a nakreslenie balónika môžeme zadať **naraz v jednom príkazovom riadku**.

### Veľa balónikov


Chceme nakresliť niekoľko farebných balónov *v ruke*. Zmažeme stránku pomocou **Znovu** a natočíme korytnačku **vľavo 30**. Do prázdneho príkazového riadka napíšeme štyri príkazy. Prvé tri nakreslí balónik. Štvrtý vráti korytnačku pomocou **vzad 120** späť *do ruky*. Potom korytnačku pootočí **vpravo 15** a opakujeme to isté.



```
? vI 30
? nechFp ? do 120 bod 60 vz 120
? vp 15
? nechFp ? do 120 bod 60 vz 120
? vp 15
? nechFp ? do 120 bod 60 vz 120
? vp 15
? nechFp ? do 120 bod 60 vz 120
? vp 15
? nechFp ? do 120 bod 60 vz 120
? vp 15
? nechFp ? do 120 bod 60 vz 120
```

Päť balónov – to je päť celkom rovnakých príkazových riadkov. Našťastie ich nemusíme znovu a znovu písať. Stlačme šípku dohora raz alebo viackrát.

Aby sme nakreslili päť balónov, **opakovali sme päťkrát** tie isté kroky. Zvolili sme ľubovoľnú farbu, nakreslili motúz a balón, cúvli vzad a otočili sa vpravo o 15 stupňov. Aj pri iných kresbách sa často opakujú tie isté kroky. Nastavme korytnačke hrúbku pera 11, ľubovoľnú farbu pera a pozrite sa, ako kreslí štvorec:



- Kolkokrát sme tieto príkazy opakovali? Štyrikrát.
- Aké príkazy sme opakovali? Príkazy **do 60** a **vp 90**.
- Čo sme nakreslili? Štvorec.

Aby korytnačka nakreslila štvorec, prikázali sme jej štyrikrát opakovať dvojicu príkazov:

**opakuj 4** do 60 vp 90

Našťastie, korytnačka slovo opakuj pozná. Môžeme jej teda v jedinom riadku prikázať:

? opakuj 4 [do 60 vp 90]

Klikni na tlačidlo **Znovu** a vyskúšaj takéto príkazy:

? opakuj 4 [nechFp ? nechHp ? do 40 vp 90]

? nechHp 30 opakuj 6 [nechFp ? do 50 vp 60]

? nechHp 1 opakuj 8 [bod 45 do 45 vl 45]



ZAPAMÄTAJME SI

- Príkaz opakuj obsahuje **počet opakovaní** a **skupinu príkazov**, ktoré chceme opakovať.
- Pred a za počet opakovaní píšeme medzeru.
- Skupina príkazov sa začína a končí hranatou zátvorkou.

ÚLOHA 1

Chceme, aby korytnačka nakreslila takéto obrázky. Doplň do príkazov opakuj chýbajúce počty opakovaní.

? nechHp 2 nechFp ?  
? opakuj [ ]  
[do 60 bod 40 vp 90]

? nechHp 7 nechFp ?  
? opakuj [ ]  
[do 50 vz 50 vl 45]

? opakuj [ ]  
[nechFp ? bod 33  
do 25 vl 30]

ÚLOHA 2

- Koľko krokov prejde korytnačka, ak jej zadáme opakuj 4 [do 60 vp 90]?
- Koľko krokov prejde, ak jej zadáme opakuj 4 [do 75 vl 90]?
- Koľko krokov prejde, ak jej zadáme opakuj 4 [do 25 vp 90 do 25]?
- Aký obrázok nakreslí?
- O koľko sa korytnačka otočí, ak jej zadáme opakuj 4 [do 30 vp 45]?

ÚLOHA 3

Chceme, aby korytnačka nakreslila takéto obrázky. Pozorne si ich prezri a doplň do opakuj chýbajúce časti.

? nechHp 2 nechFp ? vp 90  
? opakuj 4 [ ]

? nechHp 1 nechFp ?  
? opakuj 6 [ ]

? nechHp 5 nechFp ?  
? opakuj 8 [ ]

Pero hore a pero dolu

Keď sa korytnačka prechádza po stránke, kreslí svojím perom čiaru. Niekedy ju však budeme chcieť presunúť z miesta na miesto bez čiary, s vypnutým perom. Vtedy použijeme dvojicu príkazov **ph** a **pd**.

- ph** znamená **pero hore**, teda vypni pero.
- pd** znamená **pero dolu**, teda zapni pero.



## 2 OPAKUJ A POMENUJ

Nakreslime vedľa seba dva štvorce, ktoré sa nedotýkajú:

Znovu

- ? vp 90 nechHp 15 nechFp ?
- ? opakuj 4 [do 45 vp 90]
- ? ph do 70 pd
- ? opakuj 4 [do 45 vp 90]



Vždy sa zamyslime nad tým, či má korytnačka pero zapnuté alebo nie. Ak jej pero vypnem potom ho zabudneme zapnúť, budeme prekvapení, prečo na stránku nekreslí čiary. O kreslení bodov to však neplatí:

- ? znovu vp 90
- ? ph opakuj 20 [bod 9 do 15] 

### ÚLOHA 4

Pomiešali sa nám štyri obrázky a štyri programy a teraz nevieme, ktorý z nich nakreslí ktorý. Prezri si obrázky aj programy a rozhodni, čo k čomu patrí.



nechFp ? nechHp 7  
opakuj 4 [do 45 bod 50 ph do 40 pd vp 90]

nechFp ? ph  
opakuj 8 [do 50 bod 35 vz 50 v1 45]

nechHp 12 nechFp ?  
opakuj 8 [do 30 ph do 15 pd vp 45]


nechHp 11 nechFp ?  
opakuj 8 [v1 90 do 40 vp 90 do 40 v1 45 nechFp ?]

### ÚLOHA 5


- V každom z nasledujúcich riadkov je chyba. Nájdi ich.
- ? opakuj 4 do 60 v1 90
  - ? opakuj [do 20 vz 20 v1 45]
  - ? opakuj [4] [vp 90 do 80]

### ÚLOHA 6

Chceme, aby korytnačka nakreslila takéto obrázky. Pozorne si ich prezri a nájdi vhodné príkazy.




## PROJEKT CD

- Kliknime na tlačidlo  Otvoriť projekt a otvoríme projekt s názvom 2opakuj.
- Skladajme z kartičiek prekvapujúce programy a pozorujme, ako vyzerá výsledný obrázok. Použijeme rôzne počty opakovaní, rôzne farby a hrúbky pera, príkazy ph a pd, kreslenie farebných bodiek pomocou príkazu bod.

### Naučme korytnačku nový príkaz

Spomeňme si, ako sme kreslili balón. Zvolili sme hrúbku a farbu pera, nakreslili motúz a veľký bod a nakoniec cúvli späť o dĺžku motúza. Skúsme teraz korytnačke zadať namiesto týchto piatich príkazov jediný – **balón**:



```
? balón
Neviem ako sa robí balón
Buď si sa pomýlil v mene príkazu, alebo si ho ešte vôbec nezadefinoval
? |
```

Korytnačka ešte príkaz **balón** nepozná, ale rada sa ho naučí. Do príkazového riadka napíšme **uprav**, potom medzeru, úvodzovku a meno pre nový príkaz. Otvorí sa okienko s riadkami **viem balón** a **koniec**. Do riadkov medzi ne vysvetlíme korytnačke známymi príkazmi, ako sa kreslí balón.

? uprav "balón

balón	viem balón
	koniec

balón	viem balón
	nechHp 2
	nechFp ?
	do 120
	bod 60
	vz 120
	koniec

Keď dopíšeme definíciu nového príkazu, klikneme na tlačidlo **OK** a balón môžeme používať rovnako ako iné príkazy.






```
? znovu
? v1 30
? opakuj 8 [balón vp 10]
```

Naučme korytnačku ďalší príkaz a použime ho v príkazovom riadku.

? uprav "dvojBodka

balón	viem dvojBodka
dvojBodka	bod 30
	do 35
	bod 30
	vz 35
	koniec



```
? ph
? dvojBodka
? vp 90 do 35 v1 90
? dvojBodka
? vp 90 do 35 v1 90
? dvojBodka
```

```
? opakuj 4 [dvojBodka vp 90]
```

```
? pd
? dvojBodka do 35
? vp 90
? dvojBodka do 35
? dvojBodka do 35
? vp 90
? dvojBodka do 35
```

#### ZAPAMÄTAJME SI

- Nový príkaz je vlastne **nové meno**, ktoré dáme skupine známych príkazov.
- V príkaze **uprav "balón** alebo **uprav "dvojBodka** píšeme úvodzovku iba pred meno nového príkazu.

#### ČO SME SA NAUČILI

- Veľa zaujímavých obrázkov nakreslíme tak, že opakujeme tú istú skupinu príkazov niekoľkokrát. Pritom nám pomôže príkaz **opakuj**.
- Korytnačka pozná aj príkazy na zapnutie a vypnutie pera **pd** a **ph**.
- Skupine príkazov môžeme dať meno a naučiť tak korytnačku nový príkaz.
- Zatiaľ však nevieme nakresliť balóny na nitiach **premennej dĺžky**. To sa naučíme neskôr.



## STAVEBNICE PRÍKAZOV

Už vieme, že skupinu príkazov môžeme pomenovať, a tak naučiť korytnačku nové slovo. Takto sme definovali príkazy `balón` a `ďvojbodka`. Urobme teraz to isté pre `štvorec`: naučme korytnačku kresliť štvorec hrúbkou pera 7 kreslí štvorec veľkosti 25 krokov. Otvorme Imagine, opäť vytvoríme `Znovu` a napíšme:

? uprav "štvorec"

```
štvorec [viem štvorec
         nechHp 7
         opakuj 4 [do 25 up 90]
         koniec
```

Odteraz už korytnačka rozumie príkazu `štvorec` presne tak dobre, ako príkazom `ďpredu` či `do`, `vľavo`, `znovu`, `bod` a ďalším. Skúsme `štvorec` využiť na rôzne kresby. Napíšme do príkazového riadka:



? vp 90  
 ? nechFp ? štvorec do 25  
 ? nechFp ? štvorec do 25  
 ? nechFp ? štvorec do 25  
 ? nechFp ? do 20 štvorec vz 20  
 ? v1 90  
 ? nechFp ? do 20 štvorec vz 20  
 ? opakuj 8 [štvorec do 25 vz 90]

Ak chceme nakresliť vedľa seba veľa štvorcíkov, môžeme použiť príkaz `opakuj`:

? opakuj 8 [nechFp ? štvorec do 25]

Pokračuj tak, aby korytnačka nakreslila štyri vlajky. Skús to isté ešte raz, ale použi príkaz `opakuj`.

Z malých štvorcov môžeme vytvoriť aj zložité obrázky. Na zmažaní stránky použime tlačidlo `Znovu` alebo v príkazovom riadku zadáme `? znovu`.

## ÚLOHA

1

Opäť sa nám pomiešali obrázky a programy. Pozorne si ich prezri a rozhodni, čo k čomu patrí.



opakuj 8 [do 50 up 45 štvorec v1 45 vz 50 up 45]

opakuj 8 [nechFp ? štvorec up 45 do 40]

opakuj 8 [do 40 štvorec vz 40 up 45]

opakuj 8 [nechFp ? do 35 v1 45 štvorec up 45 vz 35 up 45]

## ÚLOHA

2

Ak do `opakuj` pridáme k príkazu `štvorec` aj vypnutie a zapnutie pera `ph` a `pd`, získame ďalšie zaujímavé možnosti. Chceme, aby korytnačka nakreslila takéto obrázky. Pozorne si ich prezri a nájdi vhodné príkazy.

štvorec [do 100 up 80]  
 opakuj 4 [štvorec v1 90  
 ph do 30  
 pd]



Ak by sme chceli **pozmeniť** príkaz **štvorec**, napríklad zmeniť hrúbku pera alebo veľkosť strany, vtedy znovu napíšeme **? uprav "štvorec"**

```

Štvorec      vien štvorec
                nechHp 5
                opakuj 4
                [do 40 vp 90]
                koniec
    
```

Tento štvorec už nemá stranu veľkosti 25 ale 40 a kreslí sa tenším perom.


- V menách príkazov môžeme používať dlžne a mäčkene, ale nemusíme. Pre korytnačku je **štvorec** a **stvorec** to isté slovo.
- V definícii môžeme písať niekoľko príkazov do jedného riadka. Ale aj jeden príkaz **opakuj** môže zaberat niekoľko riadkov.



Na obrázkoch z predchádzajúcej úlohy sa vždy opakuje určitý motív. Pri práci s počítačom a korytnačkou sa učíme zbadat takéto opakujúce sa motívy. Vieme, že veľa vecí okolo nás je akoby poskladaných z trojuholníkov, štvorcov, kruhov a podobných obrazcov. – motívov. Preskúmajme ich spolu.

### Trojuholník, štvorec, šesťuholník a ďalšie „uholníky“

Keď korytnačka kreslí **štvorec**, najprv nakreslí stranu, potom zahne vpravo o 90 stupňov. Túto dvojicu príkazov zopakuje štyrikrát. Pokúsme sa podobným spôsobom nakresliť **trojuholník**. Má tri strany, preto napíšme:




Znovu

```

? opakuj 3 [do 40 vp    ]
    
```


Ešte musíme doplniť, o aký uhol sa má korytnačka v každom vrchole otočiť. Skúšajme postupne rôzne hodnoty, až objavíme tú správnu – 120 stupňov.



Pri kreslení trojuholníka sa korytnačka otáča o 120 stupňov. Nie je to vlastne prekvapenie. Keď kreslí štvorec, štyrikrát sa otočí o 90 stupňov, teda spolu sa otočí o 360 stupňov – a skončí natočená rovnako ako na začiatku. Čosi podobné sa stane pri kreslení trojuholníka. Korytnačka sa trikrát otočí o 120 stupňov, teda spolu sa otočí o 360 stupňov.

#### ZAPAMÁTAJME SI

- Nakreslili sme trojuholník s tromi rovnakými stranami – teda **rovnostranný trojuholník**. O ňom vieme, že má rovnaké nielen strany, ale aj uhly, a to 60 stupňov. Aby mal **vnútorný uhol** pri vrchole 60 stupňov, musí sa korytnačka otočiť o 120 stupňov. Z matematiky vieme, že tieto dva uhly sa nazývajú susedné a spolu tvoria priamy uhol – teda 180 stupňov.






### 3 STAVEBNICE PRÍKAZOV

Naučme korytnačku nový príkaz **trojuholník**. Podobne ako so štvorcem, aj s trojuholníkom môžeme vytvárať rôzne vzory. Pred každým pokusom použijeme **Znovu** a zvolíme náhodnú farbu.

? opakuj 4 [trojuholník vl 90]  
 ? opakuj 8 [trojuholník do 40 vl 45]  
 ? opakuj 8 [trojuholník do 40 vp 45]  
 ? opakuj 8 [trojuholník ph do 45 vp 45 do 5 pd]  
 ? opakuj 8 [do 30 trojuholník vz 30 vl 45]

? uprav "trojuholník"

štvorec	viem trojuholník
trojuholník	opakuj 3 [do 40 vp 45]
	koniec



Ak správne priložíme trojuholník na štvorec, nakreslíme domček. Skúsme najprv objaviť postupnosť príkazov v príkazovom riadku. Ak sa pomýlime, nakreslíme rôzne zábavné výtvarky ako tu. Potom definujme príkaz **domček**.

domček	viem domček
štvorec	štvorec
trojuholník	do 40
	vp 30
	trojuholník
	vl 30
	uz 40
	koniec

štvorec do 40 vp 30 trojuholník vl 30 vz 40

Niekedy je vhodné pridať na koniec ešte dva kroky - **vl 30** a **vz 40**. Vďaka nim sa korytnačka vráti tam, kde začala domček kresliť. Je to dobrý zvyk. Keď sa rozhodneme nakresliť celú ulicu domov, budú sa nám domčeky jednoduchšie stavať vedľa seba.

#### ÚLOHA 3

Nauč korytnačku nový príkaz **ulica**. Je to rad desiatich domov stojacich vedľa seba.



#### ÚLOHA 4

Nauč korytnačku nový príkaz **strom**. Spomeň si na balón - od neho sa strom líši hrúbkou čiary a farbami. Použi **strom** pri ďalšom príkaze **alej**. Je to rad deviatich stromov stojacich vedľa seba.



Všimnime si, ako sme do príkazu **alej** napísali **opakuj**. Namiesto dlhého riadka sme vytvorili stĺpec, aby sa nám ľahšie čítal. To sa v príkazovom riadku nedá.


alej	viem alej
domček	opakuj 9 [strom
strom	up 90
štvorec	ph do 35 pd
trojuholník	vl 90]
ulica	koniec

### Kružnice a oblúky

Pri kreslení štvorca sme písali **opakuj 4 [do 40 vp 90]**, pri kreslení trojuholníka **opakuj 3 [do 40 vp 120]**. Ak by sme kreslili 8-uholník, napísali by sme napr. **opakuj 8 [do 15 vp 45]**. Skúsme však viac. Nie 8-uholník, ani 30-uholník či 60-uholník, ale hneď 360-uholník pomocou **opakuj 360 [do 1 vp 1]**. Takýto obrazec s 360 stranami vyzerá presne ako kružnica.

Naučme korytnačku kresliť kružnicu ako opakuj 360 [do 1 vp 1]. Spomeňme si však aj na to, ako sme kreslili ulicu. Najprv sme naučili korytnačku štvorec a trojuholník. Pomocou týchto príkazov sme ju naučili kresliť domček. Z domčekov sme napokon poskladali ulicu. Podobnú stavebnicu si môžeme urobiť aj z oblúkov. Naučme korytnačku kresliť iba štvrtinu kružnice a skúsme, čo všetko sa z nej dá vykúzlitiť.

? nechHp 4  
 ? nechFp ?  
 ? opakuj 360 [do 1 vp 1]



? opakuj 90 [do 1 vp 1]

viem štvrt  
 nechFp ?  
 opakuj 90 [do 1 vp 1]  
 koniec

? nechHp 6  
 ? opakuj 4 [štvrt]




Pravdaže, veľmi jednoduchá je kružnica.

Vytváranie nových príkazov je ako hra so stavebnicou. Zo štvrtkružnice urobme znak, lupeň alebo niečo celkom iné. Z lupeňa urobme kvet.

viem znak  
 opakuj 4 [štvrt v1 180]  
 koniec



viem lupeň  
 opakuj 2 [štvrt vp 90]  
 koniec



viem kvet  
 opakuj 8 [lupeň v1 45]  
 koniec



### Hviezdy

Naučme sa kresliť hviezdu. Kým pri štvorci sme prikázali korytnačke, aby sa otáčala 4-krát o 90 stupňov, teraz jej prikážme otáčať sa 5-krát o 144 stupňov. Zmeňme farbu stránky a nakreslime nočnú oblohu:

? nechHp 3  
 ? nechFp "žltá9"

? nechFarbaPozadia "belasá5"  
 ? opakuj 200 [hviezda do ? vp ?]

viem hviezda  
 pd  
 opakuj 5 [do 20 vp 144]  
 ph  
 koniec

### ČO SME SA NAUČILI

- Ak korytnačku naučíme nový príkaz, môžeme ho ďalej používať ako každý iný príkaz – aj v príkazovom riadku, aj v definícii ďalších nových príkazov.
- S príkazmi sa môžeme hrať ako so stavebnicou – skladať ich do väčších a väčších celkov.
- Už vieme nakresliť trojuholník, štvorec, 8-uholník... a aj 360-uholník. Ten vyzerá presne ako kružnica.

### PLÁN NA NESKÓR


Chceli by sme na oblohu nakresliť hviezdy rôznej hrúbky pera, farby a veľkosti. Príkaz hviezda však vždy kreslí hviezdu veľkosti 20, perom hrúbky 3 a farbou žltá9. Ak by sme chceli kresliť iné hviezdy, museli by sme definovať ďalšie príkazy, ako maláHviezda, veľkáHviezda, bledáHviezda... a striedavo ich používať. Ale na to, aby sme nakreslili takéto obrázky, by sme potrebovali aspoň dvadsať rôznych príkazov.



## 4 DÔLEŽITÉ UDALOSTI V ŽIVOTE KORYTNAČKY

Kliknime na korytnačku ľavým tlačidlom myši. Uvidíme, že sa nič nestane – zatiaľ sme korytnačku nenaučili, čo má vtedy urobiť. Kliknime však na ňu **pravým tlačidlom myši**. Už vieme, že to spôsob, ako sa v prostredí Imagine skúmajú vlastnosti *hráčov a vecí*, napr. korytnačky, strán, stránky a pod. Otvorí sa ponuka príkazov. Zapamätajme si, že:

- Musíme kliknúť presne na korytnačku, aby sa namiesto ponuky príkazov pre korytnačku neotvorila ponuka príkazov pre stránku.
- V prvom riadku ponuky je príkaz **Zmeň k1**. Ten budeme používať najčastejšie, pretože otvára **rodný list korytnačky**.
- V prvom riadku ponuky vidíme okrem slova **Zmeň** aj to, že sa korytnačka volá **k1**.



Zvoľme prvý príkaz ponuky. Na obrazovke sa otvorí okno s nastaveniami a informáciami o korytnačke.



### Rodný list korytnačky

Na niekoľkých **záložkách** tu nájdeme, aký má práve korytnačka tvar a smer, akú farbu a hrúbku má teraz jej pero a mnoho ďalších nastavení. Údaje o korytnačke si tu môžeme nielen **prečítať**, ale môžeme ich aj **zmeniť**.

Všimnime si najspodnejšie okienko s názvom **priKliknutí**. Je prázdne, pretože zatiaľ sme korytnačku nenaučili, ako má reagovať, keď na ňu klikneme ľavým tlačidlom myši. Napíšeme sem **do 60 v1 60**:

priKliknutí: do 60 v1 60

Potom kliknime na **OK** a rodný list sa zavrie. Klikajme teraz na korytnačku a pozorujme výsledok.



Zmažeme stránku tlačidlom **Znovu**, opäť otvoríme rodný list korytnačky a zmeňme **priKliknutí** na **nechHp ? nechFp ? do 60 v1 120**

Spomeňme si, že otáznik (pre ľubovoľnú farbu či hrúbku) môžeme použiť aj v **priKliknutí**. Zmeňme riadok **priKliknutí** na **nechHp ? nechFp ? do ? v1 ?**

### ZAPAMÄTAJME SI

- Pre korytnačku **k1** sme práve definovali **udalosť**. Každá udalosť má svoje **označenie** a **reakciu**, teda to, čo sa má stať. V našom prípade je udalosť označená slovom **priKliknutí**. Znamená to, že nastane vždy, keď na korytnačku klikneme ľavým tlačidlom myši. V prvom prípade bolo reakciou na túto udalosť **do 60 v1 60**.

Dosiaľ sme ako reakciu na udalosť **priKliknutí** používali iba základné korytnačie príkazy. V riadku **priKliknutí** však môžeme použiť aj naše vlastné príkazy ako **štvorec** alebo **strom**. Spomeňme si na príkaz **strom**. Napíšme ho ako reakciu na udalosť **priKliknutí**:

priKliknutí: strom

```
vien strom
nechFp "hnedá
nechHp 12
do 30
nechFp "olivová5
bod 40
ph vz 30 pd
koniec
```

Teraz by korytnačka pri kliknutí nakreslila strom a cúvla by tam, kde začala. Takže pri ďalšom kliknutí by nakreslila taký istý strom na tom istom mieste. Preto definujeme nový príkaz **posun**. Ako reakciu na kliknutie zadajme dva príkazy – **strom** a **posun** o 40 krokov:

priKliknutí: strom posun



? uprav "posun

```
vien posun
up 90
ph do 40 pd
vl 90
koniec
```

**ÚLOHA 1** Definuj príkaz **trojuholník**, ktorý nastaví korytnačke pero na hrúbku 5, náhodnú farbu a nakreslí trojuholník veľkosti 40. Skúmaj, aká kresba vznikne, ak udalosť **priKliknutí** definujeme takto:

- trojuholník vl 90
- trojuholník do 20 vl 30
- trojuholník do 40 vp 60

### Skok na ľubovoľnú pozíciu

Použijeme príkazy **štvorec** a **domček** z predchádzajúcej kapitoly a zmeňme udalosť **priKliknutí** na **domček posun**.



Ak chceme namiesto ulice vytvoriť osadu domov roztrúsených po stráni, použijeme nový korytnačí príkaz **nechPoz ?**, ktorý korytnačke prikazuje, aby **preskočila** na **ľubovoľnú pozíciu**.

```
vien skok
ph
nechPoz ?
pd
koniec
```

Definujeme príkaz **skok**, ktorý vypne korytnačke pero, preniesie ju na iné miesto na stránke a pero jej opäť zapne.

Zmeňme udalosť **priKliknutí** na **domček skok**.

**ZAPAMÁTAJME SI**


Príkazy, v ktorých **otáznik** vyjadruje ľubovoľnú (náhodnú) farbu, hrúbku a pôd.:

dopredu ?	vľavo ?	nechFarbaPera ?	nechPoz ?
do ?	vl ?	nechFp ?	
bod ?	vpravo ?	nechHrúbkaPera ?	
	up ?	nechHp ?	

## ÚLOHA

Zmeň udalosť **priKliknutí** tak, aby korytnačka nakreslila takéto obrázky:

2



Korytnačka si pri kliknutí zvolí náhodnú farbu, nakreslí bod 20 a so zapnutým perom skočí na náhodnú pozíciu.



Korytnačka pri kliknutí nakreslí zvláštny štvorec veľkosti 30 – každú stranu náhodnej farby a hrúbky, potom skočí inam.

## Automatické ťahanie

V programe LogoMotion **kreslíme** obrázky pomocou myši, v Imagine sme ich zatiaľ iba **programovali**. Keď spojíme kreslenie myšou a programovanie, vytvoríme zaujímavé kresby a objavíme ďalšie možnosti korytnačky.

- Kliknime na korytnačku pravým tlačidlom a v jej rodnom liste zmažeme riadok **priKliknutí**.
- Potom kliknime na záložku **Tvar**. V jej dolnej časti kliknime na štvorček pred nápisom **Automatické ťahanie**.




Pomocou **korytnačky s automatickým ťahaním** môžeme kresliť ako v kresliacom programe. Práve to je to také pohodlné. Automatické ťahanie však má iné výhody, na ktoré hneď prideme.

- Zapneme korytnačke pero – buď na prvej záložke jej rodného listu, alebo príkazom **pĎ**.
- Zvoľme farbu a hrúbku pera a ťahajme korytnačku myšou. Prvé jednoduché kreslenie máme hotové.

## Čo sa má diať pri ťahaní

Korytnačka **s automatickým ťahaním** sa stane zaujímavým nástrojom, keď pre ňu definujeme udalosť **priŤahání**. Naučme ju, aby počas ťahania myšou **opakovane** kreslila domček. Neboj sa však uistiť, že riadok s udalosťou **priKliknutí** je prázdny.



- 1 Otvorme záložku s menom **Udalosti**. Tu môžeme korytnačku naučiť, ako má reagovať na rôzne udalosti.
- 2 Kliknime na tlačidlo **Pridaj**.
- 3 V zozname udalostí zvolíme **priŤahání** a kliknime na **OK**.
- 4 Do riadka **priŤahání** napíšeme **domček** a kliknime na **OK**. Teraz ťahajme korytnačku myšou po ploche.



Tento obrázok ukazuje, že udalosť **priťahani** sa opakuje veľakrát. Korytnačka nakreslí domček vždy, keď si „uvedomí“, že ju ešte stále ťaháme.



Ak sa Ti zdá domčekov priveľa, zmeň udalosť **priťahani** na **domček čakaj 30**. Príkaz **čakaj 30** zdrží korytnačku pri každom domčeku o 30 milisekúnd.



Ak zmeníme **priťahani** na **domček čakaj 60**, medzery medzi domčekmi sa zväčšia. Skúsme aj **pd domček ph čakaj 60**.

### ÚLOHA

Skúmaj, aká kresba vznikne, ak udalosť **priťahani** zmeníme na:

3

- do ?
- nechFp ? do 50
- do 40 vp 5
- nechFp ? bodka ? čakaj 50



### ÚLOHA


4


Podľa návodu dolu definuj príkaz **kvet**. Udalosť **priťahani** zmeň na **kvet čakaj 100**. Keď budeš potom ťahať korytnačku po stránke, nakreslíš takúto farebnú lúku. Definuj príkaz **celýkvet**, ktorým korytnačka nakreslí zelenú stonku a **kvet**, potom vypne pero a cúvne tam, kde začala. Vieš na stonku pridať aj listy?

```
viem kvet
pd
nechHp 7
nechFp ?
opaku 12 [do 15 uz 15 ul 30]
nechFp "žltá"
bod 15
ph
koniec
```



### PROJEKT CD

- Klikneme na tlačidlo  **Otvoriť projekt** a otvoríme projekt s názvom **4ťahanie**.
- Experimentujme s rôznymi príkazmi, kresbami a nastaveniami a ťahaním korytnačky vytvárajme prekvapujúce kresby – ako napríklad tento *pochod lampiónov* či *trojfarebná vrtuľka*. Skúmajme a nájdime aj ďalšie možnosti.



### ČO SME SA NAUČILI

- Keď na korytnačku klikneme pravým tlačidlom, otvorí sa **ponuka príkazov**. Ak zvolíme **Zmeň k1**, ukáže sa nám korytnačkin *rodný list*. Tu môžeme nájsť a zmeniť rôzne nastavenia, napr. zapnúť alebo vypnúť pero – rovnako ako pomocou príkazov **ph** a **pd**.
- V rodnom liste môžeme definovať **udalosť priKliknutí**, čiže určiť, ako má korytnačka reagovať, keď na ňu klikneme ľavým tlačidlom myši.
- Korytnačke môžeme zapnúť **automatické ťahanie** a ťahať ju myšou po stránke.
- Okrem udalosti **priKliknutí** môžeme definovať aj iné udalosti, napríklad **priťahani**.

Už vieme, ako pomocou udalostí **priKliknutí** alebo **priŤahani** zmeniť korytnačku na **hviezda** **paličku**, ktorá dokáže na stránke rozhadzovať stromy, balóny, kvety alebo vykonať **akýkoľvek príkaz**. Teraz preskúmajme, ako si môžeme takýchto kúzelnických paličiek vyrobiť **niekoľko** a kresliť na stránke každou niečo iné.

### Ako sa rozhadzujú hviezdy

Spomeňme si na príkaz **hviezda**. Kým pri štvorci sa korytnačka otáča 4-krát o 90 stupňov, pri hviezde sa otáča 5-krát o 144 stupňov. Pomocou príkazu **nechFarbaPozadia** najprv zmeňme farbu stránky na večernú oblohu, napríklad:

```
? nechFarbaPozadia "tmavoModrá4
```

Korytnačke vypneme pero, v rodnom liste jej zapneme automatické Ťahanie a na záložke **Udalosti** definujeme **priŤahani** ako **hviezda** **čakaj 80**. Prvá kúzelnická palička je hotová.

```
priŤahani hviezda čakaj 80
```

A predsa ešte jedno vylepšenie. Ak chceme, aby nebola každá hviezda rovnako žltá, riadok **nechFp "žltá7** zmeňme na

```
nechFp ?prvok [žltá7 žltá8 žltá9 žltá10 žltá11]
```

Tento príkaz korytnačke hovorí: zmeň farbu pera tak, že si vyberieš **ľubovoľnú z týchto možností**. Hviezdy teda nakreslí rôznymi odtieňmi žltej.

```
? uprav "hviezda
```

```
vien hviezda
nechFp "žltá7
nechHp 2
pd
opaku 5 [do 15 up 144]
ph
koniec
```



### ZAPAMÁTAJME SI

- Už sme sa stretli s tromi spôsobmi, ako určovať farbu pera:

```
nechFp "modrá
nechFp ?
nechFp ?prvok [žltá modrá červená]
```

- Zadáme **meno farby** s úvodzovkou a čiarkou.
- Zadáme otáznik pre **ľubovoľnú farbu**.
- Použijeme **?prvok** a **zoznam možností**.

- V zozname možností už pred mená farieb nepíšeme úvodzvky.
- Jednotlivé možnosti v zozname neoddeľujeme čiarkou, ale medzerou.

### ÚLOHA

1

Náhodný výber zo zoznamu možností môžeme tiež použiť napr. pri voľbe hrúbky pera. Ak nechceme, aby mala každá hviezda rovnakú hrúbku, namiesto **nechFp** napíšeme:

```
nechHp ?prvok [1 2 3 4 5 6]
```


Uprav príkaz **hviezda** tak, aby si korytnačka najprv vybrala niektorý odtieň žltej, potom niektorú z hrúbok od 1 do 6, a až potom nakreslila hviezdu.





## Druhá korytnačka

Nechajme zatiaľ prvú korytnačku oddychovať na oblohe a vytvoríme druhú korytnačku. Tú potom naučíme odtláčať na stránku zábavné domčeky. Urobme takéto dva kroky:

1. Klikneme ľavým tlačidlom myši na nástroj s názvom **Nová korytnačka**. Šípka, či **kurzor myši** sa vtedy zmení na tenký obrýs korytnačky.
2. Presuňme kurzor myši niekam do stránky a klikneme. Na tomto mieste vznikne druhá korytnačka.



Skôr než novú korytnačku zapojíme do práce, naučme sa:



Každá korytnačka má svoje meno. Klikneme na korytnačku pravým tlačidlom myši a prečítajme si ho v prvom riadku ponuky. Otvorme jej rodný list, vypneme pero a zapneme **automatické ťahanie**.

Klikneme na korytnačku pravým tlačidlom a zvolíme druhý príkaz. Kurzor sa zmení na a dovoľí korytnačku rýchlo **presunúť** na iné miesto. Prenášanie ukončíme ďalším kliknutím.

Ak vytvoríme viacero korytnačiek a niektorú z nich chceme neskôr **zrušiť**, klikneme na ňu pravým tlačidlom a zvolíme príkaz **Zruš k2**. Imagine sa spýta: **Naozaj chceš zrušiť k2?** Klikneme na **Áno**.

Znovu vytvoríme druhú korytnačku a naučíme ju odtláčať na stránku veselý obrázok. Volá sa **chalupa1** a je iná, než domčeky, ktoré sme dosiaľ kreslili pomocou **dopredu**, **vľavo**, **nechFp** a podobne. Nakreslili sme ju v programe LogoMotion a uložili pod názvom **chalupa1**. Korytnačky vedú – okrem maľovania čiar a útvarov – aj odtláčať na stránku takéto obrázky pomocou príkazu **odtlačObrázok**. Za ním píšeme medzeru, úvodzovku a meno obrázka. Otvor rodný list korytnačky **k2** a zadaj pre ňu **prikliknutí** takto:

prikliknutí: odtlačObrázok "chalupa1 posun

Príkaz **posun** je náš vlastný a už sme ho predtým použili. Upravme ho tak, aby posunul korytnačku vpravo o 120 krokov, pretože približne taká široká je **chalupa1**. Pri každom kliknutí na **k2** teraz korytnačka odtlačí obrázok **chalupa1** a posunie sa vpravo o 120 krokov.

? uprav "posun

```
viem posun
  vp 90
  ph do 120 pd
  vl 90
  koniec
```





## 5 VIAC KORYTNAČIEK, VIAC MOŽNOSTÍ

### ÚLOHA

2

Spomeňme si, ako sme pri kreslení hviezdy použili prvok na voľbu jednej z rôznych farieb alebo jednej z hrúbok pera [1 2 3 4 5 6]. Ten istý trik môžeme použiť v príkaze **odtlačObrázok**. V počítači nie je medzi obrázkami uložená iba **chalupa1**, ale aj **chalupa2** a **chalupa3**. Uprav pre **k2** jej udalosť **prikliknuti** tak, aby vždy náhodne odtlačila niektorú z **chalupa1**, **chalupa2** alebo **chalupa3** a potom sa posunula vpravo.



### ÚLOHA


3

Už máme na stránke dve korytnačky – jedna kreslí hviezdy, ďalšia stavia domy. Vytvoríme tretiu korytnačku, vypni jej pero a zapni automatické ťahanie. Definuj udalosť **prikliknuti** iná **Hviezda**, v ktorom si korytnačka zvolí náhodnú žltú a nakreslí bod medzi veľkostí [1 2 3 4 5 6 7 8]. Pre novú korytnačku definuj udalosť **prikliknuti** tak, aby po oblohe rozhadzovala takéto *iné hviezdy*.




## Panel kreslenia

Ak chceme ku komínu domčeka na stránke prikresliť dym (alebo čokoľvek iné), klikneme na tlačidlo **Ukáž/Skry panel kreslenia**. Vtedy sa otvorí panel so **základnými nástrojmi na kreslenie**, aké už poznáme z programov Skicár či LogoMotion. Tu si môžeme zvoliť čiaru, obdĺžnik, kružnicu či sprej, rôzne farby a hrúbky nástrojov... a kresliť. Panel kreslenia sa skryje ďalším kliknutím na tlačidlo **Ukáž/Skry panel kreslenia**.



Ak si zvolíme bielu alebo bledosivú farbu, hrúbku 2 a ako nástroj sprej, môžeme domčekom prikresliť dym, alebo večernú dedinku zasypať snehom.

## Ulož si pozadie. Ulož si projekt



Ak sa nám stránka, ktorú sme vytvorili páči, môžeme ju **uložiť na disk ako obrázok**. Klikneme pravým tlačidlom myši do stránky – otvorí sa ponuka príkazov, ktoré jej patria. Zvoľme príkaz **Ulož pozadie** a v okne **Uložiť obrázok** zadajme **Názov súboru**.

Môžeme tiež **uložiť celý projekt**, teda uložiť stránku, korytnačky, ktoré sme postupne vytvorili, aj príkazy, ktoré sme pre ne definovali. V **Hlavnom paneli** klikneme na tlačidlo **Uložiť projekt**. V okne, ktoré sa otvorí, zadajme **Názov súboru**. Uložený projekt môžeme nabadúce opäť otvoriť kliknutím na tlačidlo **Otvoriť projekt** a ďalej ho používať, vyvíjať či meniť.

Pravdaže, okrem toho môžeme svoju stránku aj **vytlačiť na papier**. V príkaze **Súbor** hlavnej ponuky použijeme príkazy **Nastavenie tlače...** a **Tlačiť...**

### PROJEKT CD

- Klikneme na a otvoríme projekt **Svzory**.
- Urobme dvojklik na niektorú zo šiestich kartičiek a napíšme na ňu príkazy, napr. **pd do 15 vp 45 ph čakaj 50**
- Myšou chytíme červený krúžok a odťahujeme ho na niektorú korytnačku. Takto ju naučíme, čo má opakovane robiť **priťahaní**.
- Vytvárame rôzne programy na kartičkách, pripájame ich na korytnačky a tie potom ťahajme po stránke.
- Klikneme na **nový príkaz** alebo meno vlastného príkazu. Použijeme ich na kartičkách.

<p>                 cikCak                  domček                  hviezda                  posun                  štvorec                  vzor1                  vzor2                  nový príkaz             </p>													
<p>                 nová korytnačka                  zmaž kresbu                  farba pozadia                  zruš korytnačky             </p>	<table border="1"> <tr> <td>k2</td> <td>vzor1 čakaj 100</td> <td>k1</td> <td>vzor2 čakaj 100</td> </tr> <tr> <td>k3</td> <td>pd cikCak ph vp 80 čakaj 100</td> <td>k2</td> <td>vzor1 čakaj 100</td> </tr> <tr> <td>k4</td> <td>pd do 15 vp 45 ph čakaj 50</td> <td></td> <td></td> </tr> </table>	k2	vzor1 čakaj 100	k1	vzor2 čakaj 100	k3	pd cikCak ph vp 80 čakaj 100	k2	vzor1 čakaj 100	k4	pd do 15 vp 45 ph čakaj 50		
k2	vzor1 čakaj 100	k1	vzor2 čakaj 100										
k3	pd cikCak ph vp 80 čakaj 100	k2	vzor1 čakaj 100										
k4	pd do 15 vp 45 ph čakaj 50												

### ČO SME SA NAUČILI

- Príkazom **nechFarbaPozadia** môžeme zmeniť farbu stránky. Do príkazového riadka napíšme meno príkazu a stlačíme kláves **F9**. V pomôcke **Farby** si zvolíme farbu a jej jas.
- Pomocou **?prvok [ ... ]** môžeme náhodne vybrať **niektorú z možností**, napr. niektorú farbu alebo hrúbku.
- Pomocou nástroja **Nová korytnačka** vytvárame ďalšie korytnačky. Každá z nich má svoje meno, svoj rodný list a vlastné údalosti.
- **Korytnačka** môže pomocou **odtlačObrázok** opečiatkovať na stránku niektorý hotový obrázok.
- **Stránku** môžeme upraviť aj pomocou nástrojov **Panelu kreslenia**.
- **Stránku** môžeme uložiť alebo vytlačiť. Uložiť môžeme aj celý projekt.




# KORYTNAČKY A ICH TVARY

Dosiaľ sme pomocou korytnáčiek na stránku buď **kreslili**, alebo **odtláčali** hotové obrázky. Teraz však spravíme niečo celkom iné – do chalúp pridáme živé deti. Budú to korytnačky a tvar **zmeníme** na veselé tváre. Z **korytnáčiek-pastieliek** sa stanú **korytnačky-herci**.




- V predchádzajúcej kapitole sme vytvorili a uložili na disk obrázok s chalupami a hviezdami na nočnej oblohe. Teraz ho využijeme. Otvorme Imagine a pravým tlačidlom myši klikneme do ponuky príkazov zvolíme **Pozadie zo súboru...** a vyberme si svoju večernú dedinku.



Ak si obrázok v piatej kapitole neuložil na disk, musíš znovu vytvoriť.


- Otvorme rodný list **k1**, vypneme jej pero, zapneme **Automatické ťahanie** a klikneme na tlačidlo **Tvar...** Medzi obrázkami nájdeme šibalského chlapca s menom **Moricko**. Klikneme naň a tvar korytnačky sa zmení.



## ÚLOHA

1

Nezabudnime, že **k1** je naďalej **celkom obyčajná korytnačka**, iba inak vyzerá. Ak **Moricko** vykúka z ľavého okna domčeka a do príkazového riadka zadáme **vp 90 do 45 v1 90**, hlavička **preskočí** do vedľajšieho okna. Nájdi správne číslo pre príkazy **vp 90 do** a **v1 90**, aby **Moricko** vykukol z iného domu na obrázku.



Aj keď **k1** teraz nevyzerá ako obyčajná korytnačka, rozumie všetkým korytnačím príkazom:

- môžeme ju presúvať príkazmi **dopredu** alebo **vzad**,
- môžeme jej určovať smer pomocou **vpravo** a **vľavo** – samotná tvár sa však pritom nebude otáčať,
- môžeme ju skryť a znovu ukázať príkazmi **skry** a **ukáž**,
- môžeme otvoriť jej rodný list a definovať **prikliknutí**, napríklad ako **do 10 čakaj 500 vz 10**,
- môžeme buď v rodnom liste, alebo príkazom **nechTvar** zmeniť jej tvar z chlapca na dievča.

Ešte krajšie však bude, ak si deti do domčekov vytvoríme viac. Použijeme nástroj **Nová korytnačka** a vytvoríme **k2**. V jej rodnom liste vypneme jej pero, zapneme **Automatické ťahanie** a zmeňme tvar na **Jano**. Rovnako vytvoríme aj korytnačky s tvarmi **Juro**, **Dusan**, **Andula** a **Karol**.

? **nechTvar "Andula**



## ÚLOHA

2

Rôznym deťom priradíme v ich rodných listoch rôzne udalosti **prikliknutí**. Skúmame, ako sa budú pri kliknutí správať, ak budú reakcie napríklad takéto:

- opakuj 10 [do 10 čakaj 300 vp 180]
- vp 90 opakuj 10 [do 5 čakaj 200 vp 180] v1 90
- opakuj 1000 [do 1 čakaj 10 vz 1 vp ?]

## Oslovovanie korytnačiek

Teraz máme na stránke niekoľko korytnačiek. Jedna vyzerá ako **Moriccko**, ďalšia ako **Andula**, iná ako **Juro**... Ak chceme v príkazovom riadku zadať príkaz **skry** pre niektorú z nich, musíme ju správne osloviť:

? **k3'skry**

? **k1'skry k2'skry k3'ukáž**

? **k3'skry**  
meno apostrof prikaz

- Apostrof vyzerá ako čiarka na vrchu riadka.
- Ani pred ním, ani za ním nepíšeme medzeru.
- Za apostrof môžeme napísať iba jeden príkaz.

Postupne otvoríme rodné listy korytnačiek a v okienku s názvom **Meno** prepíšeme ich mená na **Moriccko**, **Andula**... Potom aj oslovenia budú krajšie:

? **Andula'skry čakaj 2000 Andula'ukáž**

### ZAPAMÄTAJME SI

Sú tri spôsoby, ako zmeniť tvar korytnačky:

**nechTvar "Andula**  
**nechTvar + kláves F9**  
**tlačidlo Tvar... v rodnom liste**


- Pred meno obrázka píšeme úvodzovky.

1. Použijeme **nechTvar** a zadáme **meno obrázka**.
2. Použijeme **nechTvar** a klávesom **F9** otvoríme **pomôcku** **Obrázky**.
3. V rodnom liste korytnačky klikneme na **tlačidlo** **Tvar...**

### Drevené kocky

Kliknime na **Súbor**, v ponuke príkazov zvolíme **Nový projekt...** a zrušíme korytnačku. Vytvoríme tlačidlo **Znovu**, ktoré pri zapnutí vykoná príkaz **znovu**. Vyrobíme si spolu jednoduchú stavebnicu z drevených kociek.

- Pomocou nástroja **Nová korytnačka** vyrobíme pri ľavom okraji stránky korytnačku. Vypustíme jej pero a zapneme **Automatické ťahanie**. Potom kliknime na tlačidlo **Tvar...** a vyberme tvar **kocka**.
- Definujme pre ňu udalosť **priĽavomHore**. Tá nastane vždy, keď kocku ťaháme myšou, a potom pustíme ľavé tlačidlo. Reakciou nech je príkaz **domov**. Korytnačka vtedy skočí **domov**.




priĽavomHore odtlač domov


Toto riešenie je lepšie: keď kocku niekam odtiahneme a pustíme ľavé tlačidlo myši, najprv tu príkazom **odtlač** odtlačí **sama seba**, a až potom skočí **domov**.

#### ZAPAMÄTAJME SI

- Príkazom **odtlačObrázok "chalupa1** korytnačka odtlačí obrázok s daným menom.
- Príkazom **odtlač** korytnačka odtlačí sama seba.



Už teraz môžeš vytvárať jednoduché stavby z červených kociek.



Obrázky s viacerými zábermi môžeme vytvárať alebo skúmať napríklad v kresliacom programe LogoMotion.

Určite by si naše stavby zaslúžili aj iné farby než červené. Niektoré tvary korytnačiek obsahujú **viacero záberov**. Napríklad **kocka** obsahuje štyri zábery so štyrmi rôznymi farbami.

- V príkazovom riadku zadajme príkaz **nechZáber** s číslom záberu:


? **nechZáber 2**

Červená kocka sa zmení na žltú. Ak zadáme **nechZáber** zmení sa na modrú. Ak zadáme **nechZáber 1**, opäť sa zmení na červenú.

- Ak v príkazovom riadku zadáme **nechZáber ?**, korytnačka si **náhodne** zvolí niektorý zo svojich záberov – bude červená, žltá, modrá alebo zelená.
- Pridajme preto do hry ďalšie tlačidlo **Zmeň farbu**, ktoré vykoná práve toto.

**ÚLOHA**  
3

- Pridaj ďalšie tlačidlo, ktoré pri zapnutí vykoná **nechZáber** ?
- Postav z kociek tieto stavby. Mysli na správne farby a poradie, v akom budeš kocky prikladať.




Teraz pridajme aj iné stavebné prvky:

Znovu  
Zmeň farbu

Dávajte pozor na to, kde každú korytnačku vytvoríme. To je totiž jej **domov**. Sem sa bude vracat po pustení ľavého tlačidla myši.

- Pomocou nástroja **Nová korytnačka** vyrobme blízko prvej kocky ďalšiu. Zmeňme jej tvar na **strecha**, vypneme jej pero a zapneme **Automatické ťahanie**. Definujme pre ňu udalosť **prilavomHore** ako **odtlač domov**.
- Rovnako vytvorme ďalšie korytnačky s tvarmi **doska**, **stlp** a **obluk**.



**viem zmeňFarby**

- k1' nechZáber ?
- k2' nechZáber ?
- k3' nechZáber ?
- k4' nechZáber ?
- k5' nechZáber ?

**koniec**


Tlačidlo **Zmeň farbu** doteraz menilo záber, čiže farbu iba prvej kocke **k1**. Definujme preto príkaz **zmeňFarby**, ktorý postupne osloví všetky korytnačky **k1, k2, k3, k4** a **k5**. Tlačidlo pri kliknutí vykoná tento príkaz.

**ÚLOHA**  
4

Zmeň tvary korytnačiek na iné obrázky – stromy, zvieratá, ľudí... a vytvor podobný projekt na pečiatkovanie. Postav potom namiesto hradu z kociek mesto alebo zoológickú záhradu.

**PROJEKT CD**

- Klikneme na a otvoríme projekt **6kocky**.
- Stavajme z drevených kociek podľa vzoru. Pozorne si premyslime poradie, v akom prikladáme každý diel.
- Potom si namáčajme štetec do farieb a drevené kocky zafarbíme presne podľa návodu.



**ČO SME SA NAUČILI**

- V rodnom liste korytnačky môžeme zmeniť jej tvar na iný obrázok. To isté spravi aj príkaz **nechTvar**.
- Pomocou príkazov **skry** a **ukáž** môžeme korytnačku skryť alebo opäť ukázať.
- Ak máme na stránke niekoľko korytnačiek, musíme vždy **osloviť** tú, ktorej zadávame príkaz.
- Definovali sme udalosť **prilavomHore** s reakciou **odtlač domov**. Korytnačka – drevená kocka vtedy **odtlačí svoj vlastný tvar** a vráti sa domov, čiže tam, kde sme ju vytvorili.
- Niektoré tvary korytnačiek obsahujú viacero záberov. Korytnačka ich môže vymieňať pomocou príkazu **nechZáber**.


V predchádzajúcej kapitole sme menili tvary korytnačiek na domčeky a tváre detí. Teraz prejdeme tvary, ktoré sa hýbu ako postavičky v kreslených filmoch. Takéto tvary nazývame **animované**.

- Začnime **Nový projekt**, v rodnom liste **k1** klikneme na **Tvar...** a vyberme obrázok **potvorka**.
- Potvorka stojí na mieste, ale už hýbe nohami a žmurká.
- Je to obyčajná korytnačka, ktorá môže kresliť, alebo sa len tak *prechádzať* po stránke.
- Ak jej povieme **do 50**, posunie sa o 50 krokov. Bolo by krajšie, keby túto vzdialenosť prešla pomaly:


? uprav "pomaly

```
viem pomaly
opakuj 50 [do 1 čakaj 30]
koniec
```

? pomaly



Každý záber animovaného tvaru obsahuje niekoľko obrázkov, ktoré sa trochu líšia a tak vyjadrujú pohyb. Celú animáciu na obrazovke za nás riadi Imagine.



Vela animovaných tvarov nájdeme priamo v prostredí Imagine, ďalšie môžeme vytvoriť v kresliacom programe LogoMotion alebo RNA. Vyrobiť perfektnú chodiacu postavičku je veľmi ťažké. Ale aj tak určite vyrobíme vela jednoduchých zábavných animácií.

### Oslovme niekoľko korytnačiek

Nástrojom **Nová korytnačka** vytvoríme na stránke niekoľko korytnačiek a zmeňme im tvary na rôzne potvorky. Definujme príkazy **oblúkL** a **oblúkP**. Ak chceme pre **k3** zvoliť hrúbku pera 5, napíšeme **k3' nechHp 5**. Ak chceme zvoliť náhodnú hrúbku, napíšeme **k3' nechHp ?**. Ak chceme spraviť to isté pre všetky korytnačky, použijeme príkaz **pre**:

? **pre všetky** [nechHp ?]

? **pre všetky** [nechFp ?]

alebo v jednom riadku

? **pre všetky** [nechHp ? nechFp ? oblúkL]

Ak chceme, aby kreslili iba niektoré korytnačky, oslovíme ich takto:

? **pre** [k1 k3 k5] [oblúkP vp 90 oblúkP]


Ak oslovujeme **jedinú korytnačku**, napíšeme jej meno, znak apostrof a príkaz. Ani pred, ani za apostrof nepíšeme medzeru.

Ak oslovujeme **niekoľko korytnačiek**, použijeme **pre**, zoznam mien a zoznam s príkazmi.

Ak oslovujeme **všetky korytnačky**, použijeme **pre všetky** a zoznam s príkazmi.

```
viem oblúkL
opakuj 90 [do 1 vl 1 čakaj 10]
koniec
```

```
viem oblúkP
opakuj 90 [do 1 up 1 čakaj 10]
koniec
```




**k2'oblúkL**

**pre** [k1 k3 k5] [oblúkL

**pre všetky** [oblúkL]

**ÚLOHA 1**

Vytvor tím športovcov, ktorí nakreslia olympijské kruhy ako synchronní plavci – teda presne rovnako a naraz. Vytvor päť potvoriek, nastav im farby a hrúbky pier, rozmiestni ich v bazéne a zadaj správny príkaz.



**Procesy – motorčeky pre korytnačky**

Začnime nový projekt a zmeňme tvar k1 na niektorú potvorku.

Korytnačku môžeme rozhybať aj tak, že príkazom **každých** pre ňu spustíme **proces**. Je to malý neviditeľný motorček, ktorý neustále vykonáva nejakú činnosť. Ak zadáme napr.:

? každých 30 [do 1 vp 1]

korytnačka začne chodiť bez prestávky dokola a my jej môžeme zadávať ďalšie príkazy. Skúsme:

? v1 90

? nechHp 10

? nechFp "modrá"

Pre tú istú korytnačku môžeme spustiť aj ďalší proces – motorček, napríklad:


? každých 1000 [nechFp ?]

? každých 1200 [nechHp ?]

ako často? čo urobiť?

**každých 50 [príkaz príkaz ... ]**

Číslo v príkaze **každých** vyjadruje v milisekundách, ako často sa má vykonať zoznam príkazov. Príkaz **každých 1000 [ ... ]** znamená každú sekundu. Každých 20 či 30 znamená veľmi často.



Všetky rozbehnuté procesy bežia až dovedy, kým neklikneme v Hlavnom paneli na tlačidlo **Zastaviť všetky procesy**.

**ÚLOHA 2**

Skúmaj nasledujúce skupiny procesov pre k1. Po každom pokuse klikni na **Zastaviť všetky procesy**.

ph každých 30 [do 1] každých 1000 [bod 25]

pd každých 30 [do 1] každých 1000 [vp 90]


ph vp 90 každých 30 [do 1] každých 200 [pd nechHp ? v1 90 do 50 vz 50 vp 90 ph]

**ÚLOHA 3**

Pridaj na plochu ďalšiu korytnačku a definuj známe príkazy dom a strom, v ktorých sa korytnačka **vráti** tam, kde kresbu začala. Pre k1 rozbehni proces na chodenie a proces na stavanie domčekov. Pre k2 začni proces na chodenie a stavanie stromov.

```
viem stavaj
k1'kazdych 30 [do 1]
k2'kazdych 30 [do 1]
k1'kazdych 1000 [dom]
k2'kazdych 900 [strom]
koniec
```

Ak je na stránke niekoľko korytnačiek, musíme osloviť tú, pre ktorú rozbiehame proces.






## Súradnice, pozície a súradnicová sústava

Každá korytnačka má na stránke svoju pozíciu. Keď sa hýbe, jej pozícia sa stále mení. Pozíciu označujeme dvojicou čísel, napríklad [60 90] alebo [100 -30]. Prvé číslo je x-ová súradnica, druhé číslo je y-ová súradnica. Pozícia [0 0] je presne uprostred obrazovky.

Zvoľme farbu pera a spustíme v príkazovom riadku „strážny“ proces, ktorý každých 50 milisekundov vráti korytnačku na pozíciu [0 0]. Zapneme korytnačky v rodnom liste Automatické ťahanie a ťahajme ju po stránke myšou:

? každých 50 [nechPoz [0 0]]



Na stránke je ukrytá súradnicová sústava. Každý bod má dve súradnice: X-ovú a Y-ovú, napríklad 10 a 50. V Logu zapisujeme body do hranatých zátvoriek bez čiarky, teda [10 50].

Ľavý obrázok vznikol tak, že sme korytnačku ťahali ako by po kružnici. Čiary do stredu stránky spôsobil strážny proces pretože korytnačku stále sťahoval späť do pozície [0 0]. Výslednú kresbu sme potom vyfarbili v kresliacom programe LogoMotion rôznymi odtieňmi oranžovej.

Keď použijeme nástroj Nová korytnačka a klikneme do stránky, vznikne tu nová korytnačka. Keď ju potom odídeme inam a zadáme príkaz domov, korytnačka sa vráti presne na pozíciu, kde vznikla.


Zastavme strážny proces a spustíme iný: **každých 1000 [nechPoz [0 0]].** Potom pridajme ďalší proces: **každých 40 [do 4 vp 1]**

Pridajme ešte jeden proces, ktorý každých 1000 milisekundov zmení hrúbku na niektorú z [1 2 3 4] a zvolí niektorý z odtieňov oranžovej.

Zastavme procesy a dva z nich zmeňme. Strážny proces nech preskočí na pozíciu [0 0] najprv príkáže bod 15. Proces na chodenie zmeňme tak, aby korytnačka išla dopredu buď o 1, 2, alebo 3.

Už vieme, že prvé číslo v pozícii korytnačky je jej x-ová súradnica a druhé jej y-ová súradnica. Vieme aj to, že presná pozícia sa dá nastaviť príkazom `nechPoz`, napr. `nechPoz [50 100]` alebo `nechPoz [0 0]`. Korytnačke však môžeme nastaviť aj zvlášť x-ovú súradnicu a y-ovú súradnicu. Ak napr. povieme `k3'nechXSúr 20`, korytnačka k3 zmení svoju x-ovú súradnicu na 20. Ak povieme `k3'nechYSúr k1'ySúr`, k3 zmení svoju y-ovú súradnicu na takú, akú práve má k1.


- Zvoľme **Nový projekt**. Vytvoríme korytnačky k2 a k3 a zmeňme ich tvar na kružok.
- Rozbehneme procesy:
  - k1'každých 20 [do 1 vp 1]
  - k2'každých 20 [nechYSúr k1'ySúr]
  - k3'každých 20 [nechXSúr k1'xSúr]



## Človeče, nehnevaj sa

Vytvoríme jednoduchú podobu známej hry *Človeče, nehnevaj sa*. Každý z dvoch hráčov bude mať za úlohu prejsť zo svojho štartu okolo celej dráhy a vojsť do svojho domčeka. Kto bude prvý, vyhrá.


- Zvoľme **Nový projekt**.
- Kliknime pravým tlačidlom do stránky a v ponuke príkazov zvoľme **Pozadie zo súboru...** Vyberme si obrázok **človece**.



- Otvorme rodný list korytnačky, premenujeme ju na **hráč1**, vypneme jej pero a zmeňme tvar na **potvorka2**.
- Pre prvého hráča vytvoríme tlačidlo **Štart** s udalosťou **hráč1'nechPoz [-275 50]** **hráč1'nechSmer 0**. Prvý hráč sa pri štarte bude pozerat hore.
- Pre prvého hráča vytvoríme tlačidlo **Otoč** s udalosťou **hráč1'vp ?prvok [90 180 270]**
- Pre prvého hráča vytvoríme tlačidlo **Choď** s udalosťou **hráč1'opakuaj 10 [do 5 čakaj 30]**
- Pomocou nástroja **Nová korytnačka** vytvoríme druhú korytnačku, premenujeme ju na **hráč2**. Vypneme jej pero a zmeňme tvar na **potvorka9**.
- Pre druhého hráča vytvoríme tlačidlo **Štart** s udalosťou **hráč2'nechPoz [275 -50]** **hráč2'nechSmer 180**. Druhý hráč sa pri štarte bude pozerat dolu.
- Pre druhého hráča vytvoríme tlačidlo **Otoč** s udalosťou **hráč2'vp ?prvok [90 180 270]**
- Pre druhého hráča vytvoríme tlačidlo **Choď** s udalosťou **hráč2'opakuaj 10 [do 5 čakaj 30]**

Hráči kliknú na svoje štartovacie tlačidlá a ďalej sa v hre striedajú.

- Ak nie je prvý hráč natočený správnym smerom, raz klikne na svoje **Otoč** a nasleduje druhý hráč.
- Ak je prvý hráč natočený správnym smerom, môže raz alebo niekoľkokrát kliknúť na svoje **Choď**. Keď už nemôže pokračovať bez otočenia, nasleduje podľa rovnakých pravidiel druhý hráč.
- Ak hráč vybehne z dráhy von, začína znovu tlačidlom **Štart**.
- Kto prvý obíde celú dráhu a vojsť do svojho domčeka?



### ČO SME SA NAUČILI

- Niektoré tvary korytnačiek sa môžu hýbať ako postavičky z animovaných filmov.
- Korytnačky môžeme oslovovať po jednej alebo po skupinách. Môžeme ich rozhybať pomocou skrytých motorčekov – procesov.
- V stránke je ukrytá súradnicová sústava. Niekedy určujeme korytnačke súradnice pomocou **nechXSúr** a **nechYSúr**, niekedy jej určujeme pozíciu pomocou **nechPoz** a smer pomocou **nechSmer**.

dopredu 50  
vpravo 45  
nechFarbaPera "modrá"  
nechHrúbkaPera 12

V príkaze **dopredu** určujeme, o koľko krokov sa má korytnačka posunúť dopredu. Ak zadáme **dopredu 10**, posunie sa o malý kúsok. Ak zadáme **dopredu 50**, posunie sa o viac. Hovoríme, že korytnačka sa príkazom **dopredu** presunie o **úsek premennej dĺžky**. Skutočnú dĺžku určuje číslo za menom príkazu. Podobne povieme, že korytnačka sa príkazom **vpravo** otočí o **uhol premennej veľkosti**. Povieme tiež, že pero korytnačky sa príkazom **nechHp** nastaví na **premennú hrúbku**.

dopredu ■ o aký úsek?  
vpravo ■ o aký uhol?  
nechFarbaPera ■ na akú farbu?  
nechHrúbkaPera ■ na akú hrúbku?

strom ■ akej výšky?  
štvorecV ■ akej veľkosti?  
štvorecFp ■ akej farby čiary?  
pomalyDopredu ■ o aký úsek?

Tu vidíme niekoľko základných príkazov spolu s ich **premennými**, ktoré určujú dĺžku úseku, uhol, farbu pera, hrúbku pera a pod.

Tu je niekoľko našich vlastných príkazov, ktoré chceme v tejto kapitole definovať tak, aby mali svoje **vlastné premenné** – výška stromu, veľkosť štvorca, farba štvorca, dĺžka úseku a pod.

Chceme definovať vlastný príkaz **strom** s **premennou výškou**. Budeme ho používať podobne ako napr. **dopredu**. Napíšeme meno príkazu, potom medzeru a číslo – teda výšku jeho kmeňa:

- strom 30 – pre strom výšky 30,
- strom 10 – pre strom výšky 10,
- strom 40 – pre strom výšky 40...

? strom 40

```
viev strom :výška
nechFp "hnedá5
nechHp 12
dopredu :výška
nechFp "olivová4
bod 50
koniec
```

```
viev strom :výška
nechFp "hnedá5
nechHp 12
dopredu :výška
nechFp "olivová4
bod 50
koniec
```

Vlastný príkaz s premennou definujeme tak, že do prvého riadka za meno príkazu pridáme **medzeru**, **dvojbodku** a **meno premennej**. Potom v riadkoch definície použijeme dvojbodku a meno premennej tam, kde má priniesť svoju skutočnú hodnotu.

Keď potom zadáme príkaz **strom 40**, skutočná hodnota 40 sa cez premennú **:výška** prenesie do definície a vykoná. Príkaz **dopredu :výška** teda v tej chvíli vlastne znamená **dopredu 40**.

**ÚLOHA 1** Definuj príkaz **značka** s **premennou výškou**. Korytnačka nech sa nakoniec vráti späť na svoj štart.

Korytnačka najprv sivou farbou a hrúbkou 6 nakreslí paličku dĺžky **:výška**. Potom sa otočí **vIavo 90** a nakreslí červený kruh hrúbky 8. Otočí sa **vpravo 90** a cúvne späť o dĺžku **:výška**.

Červený kruh môžeme nakresliť napr. pomocou **opakuuj 180 [do 1 vp 2]**.



### Premenná veľkosť, hrúbka, farba... alebo čokoľvek

Už niekoľkokrát sme definovali štvorec pomocou opakuj 4 [do 50 vp 90]. Vždy, keď sme korytnačke prikázali štvorec, nakreslila štvorec veľkosti 50. Prečo? Lebo práve toľko vyjadruje číslo v príkaze do 50. Teraz však chceme definovať štvorecV – štvorec s **premennou veľkosťou**.

<p>? štvorecV 55</p> <pre>viem štvorecV :veľkosť opakuj 4 [do :veľkosť vp 90] koniec</pre> <p>Skutočnú veľkosť sa príkaz štvorecV dozvie cez premennú :veľkosť. Korytnačka 4-krát nakreslí čiaru takejto veľkosti a vždy sa otočí vp 90.</p> <p>? nechHp 5 ? nechFp "belasá5 ? štvorecV 50 vl 120 ? štvorecV 30 vl 120 ? štvorecV 70 vl 120</p>	<p>? štvorecHp 7</p> <pre>viem štvorecHp :hrúbka nechHp :hrúbka opakuj 4 [do 50 vp 90] koniec</pre> <p>Premennou však nemusí byť vždy iba veľkosť. Príkaz štvorecHp kreslí štvorec veľkosti 50, ale jeho premennou je <b>hrúbka pera</b>.</p> <p>? nechFp "oranžová ? štvorecHp 8 vl 120 ? štvorecHp 3 vl 120 ? štvorecHp 20 vl 120</p>	<p>? štvorecFp "oranžová</p> <pre>viem štvorecFp :farba nechFp :farba opakuj 4 [do 50 vp 90] koniec</pre> <p>Príkaz štvorecFp kreslí štvorec veľkosti 50, ale jeho premennou je <b>farba pera</b>.</p> <p>? nechHp 14 ? štvorecFp "olivová5 vl 120 ? štvorecFp "purpurová7 vl 120 ? štvorecFp "žltá5 vl 120</p>
---	---	---

V kapitole 7 sme videli, že niekedy je lepšie namiesto do 50 písať opakuj 50 [do 1 čakaj 20]. Teraz pomocou **úseku premennej dĺžky** dokážeme definovať príkaz pomalyDopredu. Ten prikáže korytnačke, aby daný úsek neprešla skokom, ale pomalým posunom.

```
viem pomalyDopredu :úsek
opakuj :úsek [do 1 čakaj 20]
koniec
```

#### ZAPAMÄTAJME SI

- Náš vlastný príkaz môže alebo nemusí mať premennú. Ak ju má, v prvom riadku definície napíšeme za meno príkazu medzeru, potom dvojbodku a meno tejto premennej.
- Premenná je jedno slovo. Môže byť buď krátke ako :s, alebo :d, alebo dlhšie ako :výška, :strana, :farba a pod.
- V riadkoch definície použijeme premennú tam, kde má priniesť svoju skutočnú hodnotu. V príkaze ju môžeme použiť aj viackrát.


#### ÚLOHA

2

Definuj svoj vlastný príkaz:

- štvorecO, jeho premenná nech vyjadruje **obvod štvorca**. Napríklad štvorecO 100 by nakreslil štvorec so stranou 25.
- veža, ktorý nakreslí vežu zloženú z malých farebných štvorčekov veľkosti 25 poukladaných nad sebou. Premennou v tomto príkaze nech je počet štvorčekov vo veži.
- hviezda, ktorý nakreslí žltú hviezdu zo záverečnej úlohy v 3. kapitole. V príkaze nech si korytnačka náhodne zvolí jednu z farieb pera [žltá8 žltá9 žltá10], potom jednu z hrúbok pera [1 2 3 4] a nakoniec nakreslí hviezdu s premennou :veľkosť. Zmeň farbu stránky na belasá3 a spusti proces:

```
? kazdych 50 [pd hviezda ?prvok [5 10 15 20 25] ph do ? vp ?]
```



3

možeme použiť aj v našich vlastných príkazoch. Definuj príkaz plnýštvorec s premennou veľkosťou, ktorý nakreslí štvorec, potom korytnačke vypne pero, natočí ju a posunie **dovnútra oblasti**. Zvolí náhodnú farbu výplne a oblasť vyfarbí. Potom sa vráti späť, otočí vl 45 a zapne pero.


```
ph
vp 45
do 20
nechFv ?
vypĺň
vz 20
vl 45
pd
```

Korytnačka musí mať pred vyplňaním **vypnuté pero**. Inak by kreslila čiaru a farbu vyliala iba do nej – možno iba do jednej bodky.



### Hry s fotografiou alebo Čarovné machuľky

Digitálne fotografie sa skladajú z miliónov farebných bodiek. Otvorme v programe LogoMotion napr. fotografiu `papert`, pozrime si ju pod lupou – a budeme prekvapení. Detail zlatej reťaze obsahuje 12 524 farebných bodiek. Na nich je 8 085 rôznych farieb. Naučme korytnačku tieto farby skúmať a **rozmazávať do machuľ**.




Detail zlatej reťaze sme zväčšili najprv 2-krát, potom 4-krát, 6-krát a 10-krát. Korytnačka sa môže pohybovať po fotografii a skúmať farbu každého bodu.

Ak by bola táto fotografia namaľovaná vodovými farbami a kvapli by sme na ňu vodu, farba daného bodu by sa rozpila na machuľku.

- Začnime nový projekt. Pravým tlačidlom klikneme do stránky a zvolíme `Pozadie zo súboru...` Vyberme obrázok `papert`. Korytnačke vypneme pero.
- Definujme príkaz `machuIa` s premennou `:veIkosť`. Korytnačka sa pozrie, akú farbu má bod priamo pod ňou a pomocou príkazu `nechFp farbaBodu` si rovnako nastaví farbu pera. Potom tu nakreslí machuľku danej veľkosti.

```
viem machuIa :veIkosť
nechFp farbaBodu
bod :veIkosť
koniec
```



Profesor Seymour Papert je jedným z tých, ktorí vymysleli programovací jazyk Logo. Naučil nás, že počítač nepatrí iba dospelým, ale i deťom. Ukázal nám, že škola môže byť hrou na objavy a zároveň dobrodružstvo.

Umiestnime korytnačku do ľavého dolného rohu fotografie, otočme ju napr. `vp 30` a postupne urobme 400 machuľ (prostredný obrázok):

```
? vp 30
? opakuj 400 [machuIa 30 do 30]
```

Znovu otvorme pôvodnú fotografiu. Zadáme príkaz, ktorý urobí na náhodných pozíciách 5000 malých machuľ (obrázok vpravo):

```
? opakuj 5000 [nechPoz ? machuIa 10]
```

#### ÚLOHA 4

Nájdí aj iné digitálne fotografie a skúmaj rôzne machuľkové príkazy.

#### ČO SME SA NAUČILI

- Náš nový príkaz môže mať vlastnú **premennú**. V riadkoch definície s ňou pracujeme rovnako, ako s číslom.
- Korytnačka si môže zvoliť farbu výplne a vyplníť celú oblasť. Môže tiež zistiť, akú farbu má bod pod ňou.



Keď si zvolíme pekné pozadie a objekty (čiže korytnačky), ktoré sa budú po ňom pohybovať, ľahko vytvoríme na stránke rôzne príbehy, obrazy alebo hry. V tejto kapitole spoznáme aj ďalšie príkazy, ktoré sa na to hodia.

### Živý obraz

- Zvoľme **Nový projekt** a zrušme korytnačku **k1**.
- Kliknime pravým tlačidlom do prázdnej stránky a zvolíme príkaz **Pozadie zo súboru...**
- V ponuke pozadí si vyberme obrázok **krajina**. Stránka nášho nového projektu takto dostane vzhľad prázdnej krajiny. Postupne ju budeme zaplňať loďami, domami a balónmi.
- Zapnime nástroj **Nová korytnačka** a dolu na stránke na sýto modrej vode vytvoríme novú korytnačku. Zmeňme ju takto: vypnime jej pero a zapnime **Automatické ťahanie**, zmeňme jej smer na 90 stupňov, jej meno na **lod1** a aj tvar na **lod1**.



Týmito objektmi budeme osídľovať krajinu.

- V pravom dolnom rohu vytvoríme tlačidlo **Hýbte sa**, ktoré spustí pohyb našich objektov v krajine. Jeho udalosť nech je príkaz **hýbteSa**. Zatiaľ máme na stránke jedinú loď, preto definujeme **hýbteSa** ako:
  - Dajme lodi udalosť **prikliknutí** – nech sa otočí do protismeru, teda **vp 180**.
  - Ďalšiu loď vytvoríme takto: kliknime pravým tlačidlom myši na **lod1** a v ponuke príkazov zvolíme **Skopíruj do schránky**. Potom opäť kliknime pravým tlačidlom myši – tentoraz však do stránky a zvolíme príkaz **Prilep zo schránky**. Takto sme práve vytvorili **kópiu** prvej lode.


```
viem hýbteSa
  lod1'každých 100 [do 1]
koniec
```

Dobre si všimnime toto:

- Loď, ktorá vznikla skopírovaním prvej lode, vyzerá celkom rovnako a volá sa **lod2**.
- **lod2** sa však nehýbe. Je to logické, pretože príkaz **hýbteSa** rozhýbal iba **lod1**. Kliknime preto na tlačidlo **Zastaviť všetky procesy** a zmeňme príkaz **hýbteSa** takto:

```
viem hýbteSa
  lod1'každých 100 [do 1]
  lod2'každých 200 [do 1]
koniec
```



- Lode budú plávať rôznou rýchlosťou. Prečo? Aj **lod2** zmení pri kliknutí smer plavby.






**ÚLOHA 1**

Podobne ako lod2 vyrob kopírovaním aj ďalšie lode. Môžeš im zmeniť tvar na niektorý z lod1, lod2 či lod3. Aby sa však **všetky lode plavili po mori**, musíš upraviť aj príkaz **hýbteSa**. V príkazoch každých ... použi rôzne počty milisekúnd.



- Jednu kópiu lode vytvorme na brehu mora, premeňme ju na dom1 a zmeňme jej tvar na niektorý z dom1, dom2, dom3 či dom4.
- Kopírovaním vytvorme celý rad domov, vyberme pre ne rôzne tvary a zoradíme ich na nábreží.
- Jednu kópiu domu či lode vytvorme až na oblohe a zmeňme ju na balon1. Kopírovaním vytvorme na oblohe balóny rôznych tvarov. Do príkazu **hýbteSa** pridajme pohyb aj pre ne. Niektorý môže pomaly letieť zľava doprava, iný môže pomaly krúžiť.



```
balon3'každých 150 [do 1 vp 1] alebo
balon4'každých 200 [do 1 vp ?prvok [0 1 2]]
```

Skúsme vytvoriť ten istý živý obraz ešte raz. Teraz však budeme tvary pre objekty vyberať podľa toho, **aká je na tomto mieste farba bodu**. Začneme **Nový projekt** a do stránky vložíme pozadie zo súboru **krajina**. Zelenej korytnačke vypneme pero, zapneme **Automatické ťahanie**, zmeňme jej smer na 90 stupňov a definujeme pre ňu udalosť **prilavomHore** ako náš príkaz **rozhodniSa**.

Korytnačku teraz môžeme ťahať po stránke myšou. Keď pustíme ľavé tlačidlo, korytnačka sa pozrie, aká je na tomto mieste **farbaBodu**.

```
viem rozhodniSa
ak farbaBodu = "modrá [nechTvar ?prvok [lod1 lod2 lod3]]
ak farbaBodu = "oranžová [nechTvar ?prvok [dom1 dom2 dom3]]
ak farbaBodu = "azúrová7 [nechTvar ?prvok [balon1 balon2]]
koniec
```

Ak je **modrá**, zmení svoj tvar na niektorú z lodí. Ak je **oranžová**, zvolí si niektorý dom. Ak je **azúrová7**, zmení sa na niektorý balón. Urobme ďalšie kópie tejto korytnačky a ťahaním ich rozložíme na rôzne miesta obrazu.

```
ak farbaBodu = "modrá [ nechTvar "lod1 ]
ak [ ] [ ]
podmienka Je pravda? Ak áno, vykonaj príkaz
```

V **rozhodniSa** sme použili nový príkaz **ak**. Ten preverí **podmienku** a ak je pravdivá, vykoná **príkaz** v hranatých zátvorkách. Ak podmienka pravdivá nie je, príkaz v zátvorkách sa nevykoná.

## ÚLOHA

2

Rozlož týmto novým spôsobom po obraze lode, domy a balónky. Ak korytnačkám nezmeníš mená, budú sa volať k1, k2, k3... Pomocou kliknutia pravým tlačidlom a príkazu `Daj ma navrch` sa postaraj, aby sa lode správne prekryvali a aby balónky nelietali za iné objekty. Potom uprav príkaz `hýbtesa` a celý obraz rozhyb.


## ÚLOHA

3


Za podmienkou v príkaze ak môže nasledovať skupina viacerých príkazov v hranatých zátvorkách, nie iba jeden príkaz. Hovoríme, že **telo** príkazu ak môže obsahovať aj **niekoľko príkazov**. Do príkazu `rozhodniSa` pridaj pre lode ešte jeden príkaz: `nechSmer ?prvok [90 270]`. Vieš, čo spôsobí tento príkaz?

## Preteky potvoriek


V živom obraze sa lode a balónky hýbali po svojich dráhach a nedbali na nič iné. Aj teraz osídlieme stránku objektmi – budú to naše známe farebné potvorky. Nebudú však chodiť *iba tak*, ale budú pretekať. Začnime **Nový projekt** a zrušme korytnačku `k1`. Klikneme pravým tlačidlom myši do stránky a pomocou príkazu `Pozadie zo súboru...` zvolíme obrázok `preteky`. V druhom stĺpci zľava, asi v piatom riadku zhora vytvoríme novú korytnačku (prvého hráča), vypneme jej pero, zmeňme jej tvar na `potvorka2` a smer na 90. Postupne z nej vytvoríme hráča, ktorý bude *hádzat* hračou kockou a ťahať podľa výsledku, ktorý padne.



Pozorne umiestnime potvorku tak, aby mala stred bruška v strede svojho políčka. Potom jej nastavme nejakú farbu výplne a definujme pre ňu udalosť `priKliknutí` ako `krok`.



Príkaz `krok` bude veľmi jednoduchý. Potvorka sa pomaly posunie zo stredu tohto políčka do stredu susedného políčka a **vyplní ho svojou farbou**.  
`viem krok`  
`opakuj 30 [do 1 čakaj 10]`  
`vyplň`  
`koniec`




My však chceme viac. Keď klikneme na potvorku, mala by *hodit hračou kockou* a posunúť sa vpred o toľko políčok, aké číslo padlo na kocke.

123456

- V ľavom hornom rohu stránky vytvoríme ďalšiu korytnačku, zmeňme jej meno na `kocka` a tvar na obrázok `kocka`. Ten obsahuje šesť záberov s číslicami od 1 do 6. Ak kocke prikážeme napríklad `nechZaber 3`, ukáže záber číslo 3, teda číslicu 3. Ak jej prikážeme `nechZaber ?`, ukáže náhodne niektorý záber. Ak tento príkaz zopakujeme viackrát, bude sa zdať, že sa kocka naozaj *otáča* – a zastane na niektorom náhodnom zábere.

- Definujme preto príkaz `hod`: Fialová potvorka by mala pri kliknutí *hodit* kocku a urobiť niekoľko krokov – práve toľko, koľko ukáže kocka. Preto definujme príkaz `viacKrokov` s premennou `:N`. Udalosť `priKliknutí` zmeňme na `ťah` a ten definujme takto:


`viem hod`  
`opakuj 20 [nechZaber ? čakaj 30]`  
`koniec`  
`viem viacKrokov :N`  
`opakuj :N [krok čakaj 50]`  
`koniec`



Teraz vytvoríme celú skupinu pretekárov:

- klikneme pravým tlačidlom myši na pretekára k1, v ponuke zvolíme príkaz **Presúvaj k1** a vráťme potvorku do stredu jej prvého políčka,
- klikneme pravým tlačidlom myši do stránky a príkazom **Pozadie zo súboru...** znovu vyberme **preteky**,
- klikneme pravým tlačidlom myši na pretekára k1 a zvolíme príkaz **Skopíruj do schránky**. Potom v tom istom stĺpci pod sebou vytvoríme príkazom **Prilep zo schránky** ďalších štyroch pretekárov. Zmeňme im tvary na iné potvorky a každej zvolíme nejakú farbu výplne, napríklad príkazom **pre všetky [nechFv ?]**,
- vytvoríme tlačidlo **Ďalšie kolo**, ktoré spustí príkaz **ďalšieKolo**:

```
viem ďalsieKolo
k1'tah
k2'tah
k3'tah
k4'tah
k5'tah
koniec
```




Ako pretekári spoznajú, kto prvý dobehol do cieľa? Spravia to tak, že po každom kole skontrolujú pomocou **xSúr** svoje x-ové súradnice. Ak pretekár prekročil v smere osi x hodnotu 210, vyhráva. Preto upravme príkaz **ďalšieKolo** tak, že najprv potiahne každým hráčom, potom pre každého z nich pomocou **vyhralSi?** skontroluje, či už tento hráč nevyhrál. Príkaz **vyhralSi?** obsahuje podmienku

```
ak xSúr > 210 ...
```

Ak pretekár už prekročil túto hodnotu, od radosti sa zakrúti. Potom sa všetci pretekári vrátia pomocou **nechXSúr** na štartovú pozíciu s X-ovou súradnicou -345.

```
viem ďalsieKolo
k1'tah
k2'tah
k3'tah
k4'tah
k5'tah
k1'vyhralSi?
k2'vyhralSi?
k3'vyhralSi?
k4'vyhralSi?
k5'vyhralSi?
koniec
```



Keď sa pretekári vrátia na svoje štartové políčka, môžeme si prezrieť ich športové výkony – a začať nový pretek. Pridajme preto tlačidlo **Nový pretek**, ktoré urobí pomocou príkazu **nechPozadie "pretek"** presne to, čo sme doteraz robili kliknutím pravým tlačidlom myši do stránky voľbou **Pozadie zo súboru...**

```
viem vyhralSi?
ak xSúr > 210
[opakuaj 36 [vp 90 čakaj 20]
čakaj 500
pre [k1 k2 k3 k4 k5] [nechXSúr -345]]
koniec
```

ČO SME SA NAUČILI

- Korytnačky môžeme ľahko kopírovať, a to spolu s ich tvarom a všetkým, čo už vedía.
- Pomocou príkazu **ak** dokáže korytnačka preveriť podmienku a vykonať niektoré príkazy iba vtedy, ak je podmienka splnená. Podmienkou môže byť to, či korytnačka stojí na modrej farbe, či je jej x-ová súradnica menšia ako 100 a podobne.
- Do stránky vkladáme pozadie buď voľbou **Pozadie zo súboru...**, alebo vlastným tlačidlom, ktoré pri zapnutí vykoná príkaz jazyka **Imagine Logo** **nechPozadie "meno\_obrázka"**.



Už poznáme procesy a udalosti, vieme vyrobiť veľa korytnáčiek a zmeniť im tvar. Vieme ich rozhábať a vytvoriť živý obraz. Vieme pre ne definovať vlastné príkazy. Keď teraz spojíme všetky tieto možnosti, dokážeme skúmať zaujímavé veci, alebo vytvoriť si vlastné hry.

### Hry s písmenami



Dosiaľ sme ako iné tvary korytnáčiek používali domčeky, tváre detí, chrobáky či balóny. Teraz vytvoríme korytnačky s tvarom písmen abecedy a zahráme sa so slovami. Začnime nový projekt, korytnačke na stránke zmeňme tvar na **písmena**, vypnime jej pero a zapneme **Automatické ťahanie**.

Definujme pre korytnačku udalosť **priťahani**  
tako: odtlač **nechZáber ? čakaj 150**  
Keď ju teraz budeme tahať po stránke, odtlačí  
svoj tvar na plochu a zvolí si niektorý iný záber  
– teda niektoré iné písmeno ako svoj tvar.

Keď korytnačke len tak pre zábavu zapneme  
pero a zvolíme peknú farbu a hrúbku pera,  
vytvoríme *škôlku písmen na vychádzke*.

Príkaz **nechZáber ?** znamená: *zvoľ si niektorý záber*. Teraz skúsme niečo nové: **nechZáber záber + 1**. Tento príkaz znamená: *zvoľ si nasledujúci záber*. Zmeňme takto udalosť **priťahani** a pozorujme. Písmená sa teraz odtlačajú v poradí podľa abecedy. Všimnime si aj to, že po poslednom písmene korytnačka opäť použije prvé písmeno.

- Začnime **Nový projekt**, korytnačke zmeňme tvar na **písmena** a vypnime jej pero.
- Kopírovaním vytvoríme do riadku ešte dve takéto korytnačky a spustíme pre ne proces-motorček:  
? **pre všetky** [každých 1000 [nechZáber ?]]
- Niekedy v trojici meniacich sa písmen zbadáme správne slovo. Skúsme tieto slová *zbierať*. Definujme príkaz **písmeno**:  
**viem písmeno**  
odtlač  
vzad 60  
**koniec**
- Každé korytnačke uríme udalosť **priKliknutí** ako **písmeno**.

Ak klikneme na písmeno, odtlačí sa a posunie na ďalší „riadok“. Kolko dobrých slov sa dá takto vytvoriť?

Takéto zbieranie slov je celkom napínavé. Škoda len, že vieme hľadať iba slová s tromi písmenami. Skúsme to teda inak. Namiesto procesu, ktorý stále obmieňa tieto tri písmená, vytvoríme záso-bu ôsmich písmen, ktoré budeme môcť na pracovnú plochu pridávať a uberať. Zmeníme aj spôsob ich obmieňania.

Začnime **Nový projekt**, korytnačke opäť zmeňme tvar na písmena a vypnime jej pero. Definujme pre ňu udalosť **priKliknutí** ako príkaz **zamiešaj**. V ľavom stĺpci stránky vytvorme kopírovaním takýchto osem písmen – budú sa volať k1, k2, k3... až k8. Otvorme panel kreslenia a v pravej časti stránky nakreslime **Vyplnený obdĺžnik** farby **oranžová**. To bude naša plocha na skladanie slov. Pridajme tlačidlo **Mám slovo!**, ktoré postupne preverí korytnačku za korytnačkou. Ak sa nachádza na oranžovej ploche, odtlačí sa a cúvne o 60 krokov.



viem zamiešaj  
opakuj 30 [nechZáber ?  
čakaj 10]

koniec

Keď na písmeno klikneme, 30-krát sa zamieša a zostane tak.

Tlačidlo **Mám slovo!** urobí príkaz **mámSlovo**:

viem mámSlovo

k1'písmeno

k2'písmeno

k3'písmeno

k4'písmeno

k5'písmeno

k6'písmeno

k7'písmeno

k8'písmeno


koniec

viem písmeno

ak farbaBođu = "oranžová  
[odtlač vz 60]

koniec

Kto zostaví viac slov? Aké najdlhšie?



Vľavo je zásoba ôsmich písmen. Odtiaľto si z nich prenesme, koľko chceme na oranžovú plochu. Tu klikajme na písmená, až kým ne-nájdeime slovo. Potom použime tlačidlo **Mám slovo!** a pokračujme. Písmená môžeme znovu odložiť nabok, alebo si pribrať ďalšie.


### PROJEKT CD

- Otvorme si projekt **10písmená**. Je iba o trochu zložitejší, ako program, ktorý sme spolu práve vytvorili. V ľavom stĺpci vidíme všetky písmená abecedy, a tiež *kôpku* malých kartičiek s obrázkami. Písmená môžeme ťahať myšou.
- Ak písmeno pritiahneme na bledožltú plochu, odtlačí sa tu a vráti späť na svoje miesto. To isté môžeme urobiť aj s kartičkou. Keď ale na ňu klikneme na mieste, zmení sa na nasledujúci obrázok na *kôpke*. Pomocou programu LogoMotion môžeš na kôpku pridať aj vlastné obrázky.



### Lov na bodky

Vytvorme postupne hru, v ktorej bude naša známa potvorka chodiť, loviť červené bodky a otáčať sa podľa rôznych pravidiel správania. Na ľavú časť stránky budeme pridávať rôzne tlačidlá, v pravej časti pomocou panelu kreslenia vytvorme prázdnu *miestnosť*.



- Podobne ako pri iných prechádzkach potvoriek, aj teraz definujeme príkaz **krok**. Ten posunie potvorku o 30 bodov vpred, ale skontroluje, či už nevbegla do steny. Ak áno, cúvne o 30 bodov a otočí sa **vpravo 90**, aby išla popri stene.
- Na ľavom okraji stránky vytvoríme prvé tlačidlo s nápisom **Choď!**. Udalosťou pre toto tlačidlo bude spustenie procesu **každých 500 [krok]**.
- Zapíšme tlačidlo a pozorujme potvorku, ako prechádza prázdnu miestnosť.

```
viem krok
do 30
ak farbaBodu = "oranžová5
[vz 30 vp 90]
koniec
```



Zastaviť všetky procesy

Proces chodenia po miestnosti môžeme kedykoľvek zastaviť.

Keď spustíme proces, potvorka príde k stene a ďalej už stále chodí iba pri nej. Aby sme ju mohli lepšie riadiť, pridáme na ľavý okraj ďalšie dve tlačidlá: **Vľavo** – s udalosťou **vl 90** a **Vpravo** – s udalosťou **vp 90**.


Otvorme znovu panel kreslenia a prikreslíme ďalšie steny vnútri miestnosti. Dokážeš pomocou tlačidiel **Vľavo** a **Vpravo** riadiť potvorku tak, aby nazrela do každej časti tohto labyrintu?

Pridajme ešte tlačidlo **Domov**, ktoré vráti potvorku domov (môže sa nám občas hodiť). Teraz by sme chceli v miestnosti *rozsypané* niekoľko desiatok bodiek troch farieb – modré, červené a zelené. Ako to spravíme? Na ľavý okraj medzi tlačidlá prirobíme ďalšiu korytnačku, zmeníme jej meno na **kocka** a jej tvar na obrázok zo súboru **kocka**. Príkaz **hoď** už poznáme, teraz v ňom však vynecháme **čakaj 30**, pretože potvorka bude kockou hádzať 1000-krát! Príkaz **vyrobBodku** použije **hoď**. Ak padne šestka, nakreslí niektorou farbou bodku. Príkaz **vyrobBodky** posunie potvorku po miestnosti 1000-krát. Ak je na bielej ploche, pokúsi sa tu vyrobiť bodku. Potom spraví krok a náhodne sa otočí vpravo, cez stenu však nikdy neprejde.

```
viem hoď
opakuaj 20 [nechZáber ?]
koniec
viem vyrobBodku
nechFp ?prvok [červená modrá zelená]
kocka'hoď
ak kocka'záber = 6 [bod 20]
koniec
viem vyrobBodky
opakuaj 1000
[ak farbaBodu = "biela [vyrobBodku]
krok
vp ?prvok [0 0 0 90 180 270]]
koniec
```

Ak použijeme tri nuly v zozname možností pre **vp**, zvýšime pravdepodobnosť, že potvorka pôjde rovno.

Pod **Vľavo** a **Vpravo** pridajme ďalšie tlačidlo s nápisom **Vyrob bodky**, ktoré vykoná príkaz **vyrobBodky**.




**ÚLOHA 1**

Možno sa ti nepáči, ako potvorka rozhodila bodky. Možno chceš steny v labyrinte namaľovať inak. Môžeš buď znovu otvoriť panel kreslenia, alebo v ľavom dolnom rohu steny vytvoriť ďalšiu korytnačku (bude sa volať k2). Nastav jej hrúbku pera 30, farbu pera oranžová5 a na záložke Tvar jej vypni voľbu vidno. Na stránku pod tlačidlo Vytvor bodky pridaj tlačidlo Znovu, ktoré spraví znovu k2 obdĺžnik 430 600. Definuj príkaz obdĺžnik. Vieš pridať aj príkazy na kreslenie vnútorných stien labyrintu?



Zmeňme príkaz `krok` tak, aby potvorka zbierala červené bodky. Pri svojej potulke miestnosťou nech skúma, či sa neocitla na **červenej bodke**. Ak áno, **vyleje** na toto miesto bielu farbu – teda bodku vlastne zmaže.


```
viem krok
  ak farbaBodu = "červená
    [nechFp "biela vyplň"]
  do 30
    ak farbaBodu = "oranžová5
      [vz 30 vp 90]
koniec
```



Dokážeme potvorku riadiť tlačidlami `Vľavo` a `Vpravo` tak, aby pozbierala všetky červené bodky?

**ÚLOHA 2**

Ak sa ti zdajú farebné bodky nudné, nahraď ich rozhadzovaním malých kvetov. Aj potvorka sa určite poteší! Kvety môžu mať krátke farebné lupienky, ale ich stred musí byť jednofarebný – červený, zelený alebo modrý. Zmeň príkaz `krok` tak, aby potvorka potom zbierala iba kvety s červeným stredom. Ako potom potvorka takýto kvet zmaže?



Urobme ešte jednu úpravu: červené bodky nech potvorka zbiera, na zelených nech sa otáča `vpravo 90`, na modrých `vľavo 90`. Pozorujme, ako sa zmení jej správanie. Riadme potvorku tak, aby čo najrýchlejšie vyzbierala všetky červené bodky.

```
viem krok
  ak farbaBodu = "červená [nechFp "biela vyplň"]
  ak farbaBodu = "zelená [vp 90]
  ak farbaBodu = "modrá [vľ 90]
  do 30
    ak farbaBodu = "oranžová5 [vz 30 vp 90]
koniec
```

**ČO SME SA NAUČILI**

- Keď zmeníme korytnačky na písmená, môžeme vymýšľať zaujímavé aktivity. Pritom využijeme programovací jazyk Imagine na objavovanie vlastného jazyka.
- Už sme viackrát nastavovali **náhodný záber** pomocou `nechZáber ?`. Teraz sme použili príkaz `nechZáber záber + 1` na to, aby nám korytnačka či písmeno ukázalo **nasledujúci záber**.
- Použili sme príkaz `ak` pri rozhodovaní, kam má potvorka ísť a čo má urobiť. Pomocou `ak` a hádzania kocky sme vyrobili izbu plnú farebných bodiek.
- Pri programovaní lovu na bodky nám pomohol **panel kreslenia**. Dokreslili sme steny labyrintu, v ktorom potom potvorka vyhľadáva červené bodky podľa nášho programu.



# ČO SME SA NAUČILI

Korytnačka je malý kresliaci robot, ktorý žije v programe Imagine. Tu ho môžeme riadiť príkazmi, môžeme ním kresliť obrázky, alebo mu zmeniť tvar na auto, loď či balón. Môžeme vyrobiť ďalšie korytnačky, lode, balóny... a vytvoriť vlastnú hru alebo živý obrázok.

Pomôcka – stlač F9

## Korytnačka sa pohybuje po stránke

Týmito príkazmi sa korytnačka pohybuje dopredu a vzad a otáča sa vpravo a vľavo. Ak jej vypneš pero, nebude pritom kresliť čiaru.

ph – pero hore      pd – pero dolu

**dopredu** 100

dopredu 50  
vzad 80  
vpravo 90  
vľavo 45  
dlhé slová

**do** 100

do 50  
vz 80  
vp 90  
vl 45  
krátke slová

## Korytnačka má pero a kreslí

Pero korytnačky má farbu a hrúbku.

nechHrúbkaPera 10 hrubé pero  
nechHp 15 hrubé pero  
nechHp 1 najtenšie pero  
nechFarbaPera "modrá  
nechFp "modrá



nechFarbaPozadia "belasá3    nechFarbaVýplne "žltá    vyplň  
nechFp "modrá9    bod 5    bod 20    bod 25  
modrá farba s jasom 9    Farbou svojho pera nakresli bod veľkosti 5...  
farbaBodu – aká je farba bodu na stránke pod korytnačkou?

## Náhodné farby, hrúbky, zatáčania...

do ?      vz ?      vp ?      vl ?      nechFp ?      bod ?      nechPoz ?      nechZáber ?  
posuň sa dopredu alebo      otoč sa vpravo alebo      bod náhodnej      náhodná      náhodný záber ?  
vzad, o koľko chceš      vľavo, o koľko chceš      veľkosti      pozícia      v tvare  
náhodná farba a hrúbka

## Tvoje vlastné tlačidlo

Použi nástroj Nové tlačidlo. Vytvor na stránke nové tlačidlo a klikni naň pravým tlačidlom myši. V okne vyplň udalosť prizapnutí – napr. znovu .

## Rodný list korytnačky

Klikni na korytnačku pravým tlačidlom myši a zvol príkaz zmeň k1. Otvorí sa jej rodný list. V ňom môžeš zmeniť meno korytnačky, jej tvar, udalosti a ďalšie nastavenia.

## Definuj vlastný príkaz

Pomocou uprav "balón" otvor okienko na definovanie vlastného príkazu. Medzi riadky viem balón a koniec definuj, ako sa kreslí balón. Príkaz môže mať vlastnú premennú, napr. :výška.

## Udalosti a Automatické ťahanie

Pre korytnačku definuj, ako má reagovať pri kliknutí, pri ťahaní myšou... čiže pri udalostiach prikliknutí, priťahaní alebo priLavomHore a pod. Na záložke Tvar jej zapni Automatické ťahanie.

## Viac korytnačiek

Použi nástroj Nová korytnačka a vytvor na stránke ďalšie korytnačky. Definuj pre ne ich vlastné udalosti prikliknutí alebo priťahaní... Zmeň im tvar. Správne ich oslov:

k2'skry alebo pre všetky [do 50] alebo pre [k1 k2] [domov].

opakuj 4 [do 60 vp 90]  
nechFp ?prvok [modrá žltá]  
ak xSur > 200 [vp 180]  
odtlačObrázok "chalupa1  
odtlač      odtlač svoj vlastný tvar

## Tvar korytnačky a animácie

nechTvar "Andula      nechZáber 3  
ukáž a skry      nechZáber záber + 1

## Procesy sú motorčeky

každých 50 [do 2 vp 1]  
každých 1000 [nechFp ? nechHp ?]

## Presná pozícia a presný smer

Pomocou nechPoz [100 50] nastav korytnačke presnú pozíciu. Pomocou nechXSúr -200 alebo nechYSúr 50 nastav presné súradnice. Pomocou nechSmer 90 nastav korytnačke presný smer. Pomocou domov presuň korytnačku tam, kde vznikla.



Zastaviť všetky procesy



